

ベーシックSK III

SK-1100専用

セガ・キーボード

SK-1100専用

ベーシック SK III

テキスト



SEGA®

目 次

はじめに	比較演算子表	52
第1章 コンピュータの扱い方	GOSUB, RETURN	53
本体の扱い方	ON GOTO	55
キーボードの使い方	CURSOR	56
特殊キー	ON GOSUB	59
コントロールコード	READ, DATA, RESTORE	60
第2章 コンピュータを使う	DIM (配列)	62
ダイレクト・モード (直接命令)	ERASE	66
PRINT	DELETE	66
四則演算	AUTO	67
演算子表	RENUM	68
コンマ, セミコロンの	SAVE, LOAD, VERIFY	69
第3章 プログラムの作り方	REM	71
LET, 変数	CONSOLE	72
文字変数	第4章 関数	74
CLS, LIST, NEW	RND	74
INPUT, GOTO	INT	75
END STOP	DEF FN	77
FOR ~ NEXT ~, STEP	文字列関数	78
IF ~ THEN	ASC	78


CHR \$	79
LEFT \$, RIGHT \$, MID \$	80
LEN	81
STR \$, VAL	83
TIME \$	84
SPC, TAB	85
INKEY \$	86
FRE	88
プリンター制御命令	88
OUT	89
POKE	89
PEEK	90
CALL	92
第5章 グラフィック	94
COLOR	94
SCREEN	96
LINE	97
BLINE	99
PAINT	100
PSET	101
PRESET	102
POSITION	102
CIRCLE	105

BCIRCLE	107
PATTERN, MAG, SPRITE	108
第6章 関数その2	118
三角関数(SIN, 他)	120
SGN	124
対数	125
平方根	127
HEX \$	128
SOUND	129
BEEP	132
INP	133
STICK	134
STRIG	134
VPOKE	138
VPEEK	143
付 録	
変数, 配列	144
定数	145
キャラクターセット	147
キャラクターコード	148
コマンド, ステートメント表	150
エラーメッセージ	155
サンプルプログラム	159

はじめに

コンピュータは、もう誰れでも簡単に扱える時代になって来ました。しかし、コンピュータはなんだろうか、言葉を知っていても何をするものか、よく判らない方もおられると思います。

ステレオと比べてみましょう。



図のような仕組みになっていて、外部よりプログラムという情報を入れますと、結果が TV 画面に映されます。

プログラム用の言語は何種類かありますがここでは、BASIC（ベーシック）言語を使ってみましょう。

コンピュータを扱うには、すでに出来上っているプログラムを使う事も含まれます。しかし、自分専用のプログラムを使いたいのならプログラム用言語を覚えなければなりません。

コンピュータ用プログラム言語は、用途により沢山開発されております。その中でも、初めてコンピュータを使う人にも理解されやすく、一般的な BASIC 言語を使ってプログラムを作ってみましょう。

BASIC 言語を使って出来る事は、

- 1) 計算
 - 2) 文章やデータのファイル
 - 3) 図形や絵を描く
 - 4) ゲームや音楽を楽しむ
- その他、色々な事が出来ます。

この便利なコンピュータを自在に扱う為に、BASIC 言語をマスターして下さい。言語と言うと大変難しく聞えますが、命令語の量は決して多くありません。

プログラムを構成する基本的な命令文は、案外少ないのです。まず、基本になる文を10語程覚え、あとは、テキストを見ながら、少しずつ覚えれば、自然に全部覚えてしまいます。

まずテキストを見ながら、キーをたたいて下さい。きっとすぐエラーが出るでしょう。しかし、恐れずにキーをたたいて下さい。

やがて、コンピュータは、あなたの一番忠実な友達になってくれます。

第 1 章 コンピュータの扱い方

SK-1100の扱い方

まず、SK-1100 本体と一諸に入っている取扱い説明書を読んでください。

1. まず、SG-1000を用意してください。まだ電源は入れないでください。
2. SK-1100 本体より出ているフラットケーブルを、SG-1000の差し込み口にあるエッジコネクターに、しっかりと差し込んでください。
コネクターを差し込むとき、オモテとウラを間違えないよう注意してください。
3. 接続が終了したら、SG-1000に BASIC カートリッジを差し込んでください。
(ゲームと同じです。)
4. 全部の接続が終わったら、SG-1000の電源を入れてください。

画面に、SC-3000 BASIC と表示されます。

これで、SG-1000が、SC-3000と同じ高性能コンピュータになりました。

では、このテキスト・ブックを参考にして楽しい BASIC の世界で遊んでみましょう。

SK-1100とBASIC

SG-1000にはキーボードがありませんので、パソコンとして使えませんでした。このSK-1100キーボードと、専用BASICを使って高性能パソコンとして使っていただけます。

この BASIC レベルⅢは、パソコンでどのような事が出来るのか実際に触っていただき、理解していただくためのものです。パソコンというと、大変むずかしい機械のように思っているでしょうが、決してそんな事はありません。

あなたの命令通りに動作してくれます。しかしパソコンにも理解出来る言葉と、理解出来ない言葉があります。

このテキストには、パソコンに命令する言葉が説明されております。毎日少しずつ使うことであなたもパソコンを自由に扱えるようになります。

あなたもパソコンの楽しい世界に入ってみましょう。

```

10 SCREEN 2,2 [CR]
20 CLS [CR]
30 LINE (80,50)-(180,100), C, BF [CR]
40 C=C+1 [CR]
50 IF C=16 THEN C=0 [CR]
60 GOTO 20 [CR]
RUN [CR]

```

(0 は一番上の列の0を押して下さい。それが数字の0です。)

1 行打ち終わったら [CR] キーを押して下さい。入力したプログラムがコンピュータに記憶されカーソルが 1 段下ります。

キーの使い方がむずかしいかも知れませんが、一字ずつゆっくりと打ってみてください。

キーボードの使い方

キーボードには、英文字、数字、カナ文字、記号の書き込まれたキーがあります。

1 つのキーには 4 つの文字あるいは表示があり

例




キーの配列と間隔は、英文タイプライターと同じですから、両手で楽に打てます。

まずキーを打ってみましょう。




のキーを打って下さい。



1 2 と画面に出て来ましたね。そのままキーを打つと、キーに書かれた文字や記号の内で、左下の文字や記号が画面に出ます。

それ以外の文字や記号を画面に出すには、**SHIFT** (シフト) キーや、**GRAPH** (グラフィック) キー、**カナ** キーを使います。




SHIFT


キーを押し


を押すと ! が画面に出ます。

キーの組合せ



カーソル



画面左上に  が点滅しています。これは、カーソルと言って、キーからの入力された文字や記号が表示される位置を示しています。



カーソルを移動させるには、キーボード右側の矢印のある、四個のキーで移動させます。

カーソルは、各モードで形が異なります。

英数モード 

カナモード 





グラフモード *

キーボード左下にある  、  のキーを押して下さい。カーソルが変わります。元

に戻すには、もう一度同じキーを押すと、カーソルは  に戻ります。

キートップに書かれた文字、記号を画面に出すには6つの方法があります。

／ SHIFT キーを押しながら ＼

② SHIFT 英 ⁺ 数		GRAPH ⑤	SHIFT 英 ⁺ 数		GRAPH
① 英 数		カ ナ ③	英 数		SHIFT + カ ナ ④
⑥ SHIFT + GRAPH		GRAPH	SHIFT + 英 数		SHIFT + カ ナ
英 数		カ ナ	英 数		カ ナ

SHIFT (シフトキー)

左右の **SHIFT** キーは同じ働きをします。

SHIFT キーを押したまま、数字のあるキーを打つと記号が入力されます。

SHIFT キーを押したまま、英文字のキーを打つと、英文字が小文字で入力されます。

GRAPH (グラフィックキー)

グラフィック記号を入力するためのものです。カーソルは*に変わります。

グラフィック記号は、**SHIFT** キーと併せて使えます。

**英力
数ナ**

カナ文字を入力する場合に使います。カーソルは、■に変わります。

カナ文字には、アイウエオヤユヨツのように、小さな文字も必要です。その他に、「」 °、。
ーの記号も使います。これらの文字や記号は、**SHIFT** キーと併せて使います。

SPACE (スペースキー：文字や記号の間の空間をあけます。)

A B C

スペースキーを1回押すと、
カーソルが移動して一文字分
の空間をあける。

一番下段の長いキーは、スペースキーで、文字や記号の間に、スペースを入力します。
コンピュータの場合、空間も文字と同じように一つの情報として扱われます。

今までのキーを使って、色々な文字を入力して下さい。

特殊キー

今までにキーで入力した文字や記号で、画面が一杯になりましたね。必要のない文字や、記号をいつまでも画面に出しておくのは邪魔です。では、すべて消して見ましょう。

使用するキーは

HOME/CLR (ホーム / クリア)

このキーを押すと、画面上の文字が消えて、カーソルが左上のホームポジションに戻ります。
画面上の文字、記号等がすべて不必要な時に使ってください。
キーを打ち始める前に使ってください。

SHIFT キーを押しながら、このキーを押すと、文字は消えずにカーソルだけが、ホームポジションに戻ります。最初の文字、記号等を修正するときに使います。

今までに入力した文字や記号は、プログラムとしてコンピュータの内部には、記憶されています。

プログラムとして記憶させるには、行番号又は、文番号と呼ばれるものが必要です。

行番号とは、命令文を入れた場所の番地のようなものです。
アドレス（番地）とも言われます。

CR（キャリッジ・リターン）又は（リターン）

コンピュータの場合は、キーで画面に文字を入力しても、メモリーにはまだ記憶されておられません。**CR** キーを押して、初めて記憶されます。

何か文字を入力して **CR** を押して下さい。

Syntax Error と出ましたね。（シンタックスエラー）

コンピュータに入力する場合、一定のきまりで入力しなければなりません。これをコンピュータの文法と言います。文法が間違っているとエラーになります。

文法といっても、国語や、英文法にくらべると、決してむずかしくありません。

エラーについては、エラーメッセージ表を参照して下さい。


INS/DEL (インサート/デリート)

INS/DEL キーは、文字を一文字ずつ消したり追加する場合に使います。

INS (インサート) は、文字を追加する意味です。

DEL (デリート) は、文字を消す意味です。

A B C D と入力する所を、**A B C E** と打ってしまった場合、**INS/DEL** キーを押すと、カーソルが一文字分後にさがり、E の文字を消します。そこでもう一度**D**を入力しますと、**A B C D**と直りましたね。




こんどは、A B C Dの間に何か文字を入れてみましょう。

カーソルをA B C DのCの上に持って来て下さい。

SHIFT キーを押したまま、**INS/DEL** のキーを押して下さい。カーソルの点滅が早くなりましたね。何か文字を入力して下さい。

Bの次に、今入力した文字が入り、C Dが右側に一つずれましたね。



インサートモードの時（カーソルが早く点滅している間）は、何文字でも受付けます。

インサートモードから抜け出す場合は、

- ① **CR** キーを押す（一般的に良く使うキーです）
- ② カーソルキーを押す
- ③ **SHIFT** + **INS/DEL** キーを押す

いずれかで、元の状態に戻ります。

プログラムを修正した場合、**CR** キーを押さないと、メモリー内部は書き替えられません。
実際に、キーを操作して確かめて下さい。41 ページの LIST の項を参照

FUNC（ファンクション）

いつも使う言語を記憶しているキーです。

GOTO



キーの上方に GOTO と書いてありますね。それぞれのキーにも英文が書いてあります。これは、BASIC で使われる命令文です。

FUNC キーを押したまま、どれかキーを打って下さい。キーの上方に書かれている命令文が入力されます。

これは、プログラムを作るのに便利です。

↵/BREAK (画面切替 / ブレーク)

プログラムを実行中に、プログラムを停止させたい時に使います。

SHIFT キーを押したまま、**↵/BREAK** キーを押しますと、画面が変わります。これは、このコンピュータが、2つの画面を持っているので切替えて見るためです。
一つの画面はテキスト画面で、プログラムを書く画面です。
もう一つの画面は、グラフィック画面で、グラフィックを表示する画面です。

↵/BREAK

↵ ➡ 画面を切替える時に使います

SHIFT キーを押したまま使ってください。

これは、コンピュータが、2つの画面を持っているからです。

BREAK ➡ プログラムを実行中にプログラムの実行を停止させたい時に使います。

CTRL (コントロール)

CTRL キーを押したまま、コントロールキーの表にある文字キーを打ちますと、説明にある動作が行われます。

コントロールコード

キー操作	PRINT CHR\$(値);	機 能
CTRL + A	PRINT CHR\$(1);	NULL 文字なし
C	——	BREAK プログラムの実行中止
E	5	カーソル以降の文字をクリア
G	7	BELL ビップと音を出す
H	8	DEL 文字をデリート
I	9	HT 水平 TAB
J	10	LF ラインフィード
K	11	HM カーソルを左上端に戻す
L	12	CLR 画面のクリア
M	13	CR キャリッジリターン
N	14	カナ ↔ 英数字切りかえ
O	15	() 画面をテキスト ↔ グラフィック切りかえ

キー操作	PRINT CHR\$(値);	機 能
P	16	標準文字サイズ
Q	17	横 2 倍文字サイズ (SCREEN 2)
R	18	INS インサート
S	19	キー入力 (A~Z) シフトなし大文字
T	20	" (a~z) シフトなし小文字
U	21	ラインをクリアしカーソルを左端に戻す。
V	22	ノーマルモード
W	23	GRAPH キー入力グラフモード ↔ 英字切換
X	24	クリック音の ON ↔ OFF 切換
	28	⇔ カーソル移動
	29	← "
	30	↑ "
	31	↓ "

プログラム中でコントロールコードを使う場合は、
PRINT CHR\$(値)で入力して下さい。

RESET (リセット) (リセット) SG-1000 のホールドボタンがリセットキーになります。

プログラムを実行中、又は、画面に異常が出た場合、**RESET** キーを押しますと 2～3 秒後に、電源を入れた時と同じ状態の画面に戻ります。

このキーを押すと、その時点で行っている処理が中断され、まだ使用していないメモリーの大きさが表示されます。

××× Bytes Free

このキーを押しても、入力したプログラムは消えてしまう事はありません。

以上で、キーの説明は終わりました。もう一度、キーを使って次のプログラムを打ち込んでみましょう。

10 SCREEN 2,2:CLS **CR** キー

「線を引く X座標 Y座標 X座標 Y座標 色番号

20 LINE (50,50)-(150,150),5 **CR** キー

90 GOTO 90 **CR** キー

RUN **CR** キー RUNはプログラムを実行させる命令です。

こんどは色番号の後に、Bを入れて下さい。

箱を描きます。
5, B CR

箱を塗りつぶす。
5, B F


このようにBASICで、カラーを使って線や箱が簡単に描けます。

次ページの図を見てください。

グラフィックモードの座標

X 軸 0 ~ 255

Y 軸 0 ~ 191



座標の数字を変更すると、線の引き方が変わります。

数値や文字を変える場合は、カーソルキーを使って書き替えてから **CR** キーを押してください。

第2章 コンピュータを使う


ダイレクト・モード（直接命令）

コンピュータに直接仕事をさせる方法です。

まず文字を書いてみましょう。

HOME/CLR **B** **A** **S** **I** **C** **SPACE** **T** **E** **S** **T** **CR**

画面には




と出ます。

エラーとなったのは、コンピュータが入力した命令が理解出来ないからです。

今コンピュータに、BASIC TEST と、書かせたいのですね。その場合は、コンピュータに、入力した文字を「書け」と命令しましょう。

「書け」という命令は、PRINT です。



と入力して下さい。

PRINT 文は、画面に表示させる命令文です。

画面には、

```
PRINT "BASIC TEST" ← キーで入力した文字
BASIC TEST ← PRINT 文で出力した文字
READY

```

今度は、エラーになりませんね。正しく入力されたからです。

画面に、文字や記号を書きたい時には、PRINT 文を使います。

PRINT 文で、コンピュータに文字や記号を書かせる場合は必ず、" (ダブルクォーテーションマーク) を、文字の両側に付けなければなりません。(又は引用符といいます) これは、" " で囲んだ文字や記号を、一つの固まりとして扱っているからです。

数式の場合は必要ありません。

今度は、自分の名前を書いて下さい。英文字や、カナ文字を使って下さい。

" " の間には、文字の他、数字や記号、グラフィック記号も入れられます。

" " の中に、空間を置くと文字と同じように扱われます。

→スペースキーを1回押す。(この場合は5回押す)

P R I N T " [] [] [] [] [] B A S I C [] [] [] [] T E S T " [CR]

[] [] [] [] [] B A S I C [] [] [] [] T E S T

READY

■

←一文字分のスペースの意味

このように、PRINT 文を使わないと、コンピュータは入力された結果を見せてくれません。

PRINT 文は、プログラム上で大変よく使われる命令文です。

こんどは、少し違う使い方をします。

[?] [2] [+] **[4] [CR]**

? は PRINT の代りに使えます。

6

READY



コンピュータに、 $2 + 4$ の計算をして、答を PRINT するように命令しました。

計算の結果も PRINT 文で表示させます。

[?] の記号は、PRINT 文の省略形で、どんなコンピュータでも同じです。

コンピュータに計算をさせる。

四 則 演 算

コンピュータで計算をする場合、計算に使う記号が、普通の計算と違う記号があります。

呼び名

たし算	+	プラス	[+]
ひき算	-	マイナス	[-]
かけ算	*	アスタリスク	[X]

わり算 / スラッシュ \div
 べき乗 ^ X^n

計算に使う記号の事を演算子と呼びます。

演算子と違いますが、数式に使う()カッコがあります。

数値(数の大きさ)を比べる、比較演算子というのがあります。

以上をまとめたのが次の表です。使う前によく見ておいて下さい。

演算子表

	記号	使用できる場所		内 容	優先順位
		数値式	文字式		
算 術 演 算	^	○	×	べき乗 (0^0は1)	1
	+	○	×	符号 +	2
	-	○	×	符号 -	
	*	○	×	乗算	3
	/	○	×	除算	
	MOD	○	×	余り	4
	+	○	○	加算 (文字式の場合は文字の結合)	5
	-	○	×	減算	

- カッコ () が最も優先度が高い。
- 同じ優先順位の演算子が2つ以上使われた時は、左側の演算を優先する。
- 加算記号 “+” を文字式に用いた時は、結合をあらわす。

(例) “AB” + “C” \Rightarrow “ABC”


算術演算は10進で行っている。→例えば0.01きざみの値でもけたおちがない。

論理演算は2進で行っている。

算術演算は10進12けた計算、11けた表示。

キーボードで注意する事があります。キーボードの右下の方に $\boxed{\div}$ の記号がありますね。これは、グラフィック記号ですから、計算には使えません。

計算をもう一度やってみましょう。



PRINT 5*6

30

↑
└ 答が、正の数の場合は、+記号が省略されるので一文字空く。

```
PRINT 10 / 3
3.3333333333
```

PRINT は **FUNC** キーと



を使っても便利です。

```
? 3 ^ 4 CR ( 3 × 3 × 3 × 3 ) の意味      ? INT ( ( 3 ^ 4 ) + 0.1 )
80.9999999999                               81
```

演算機能は、精度の高い10進計算11桁表示で行っております。

```
? 1000000 * 10000 CR
100000000000                                11桁
```

```
? 10 / 3
3.3333333333 CR      少数点以下 10 桁
```

これより大きい数字や、小さい数字の場合は、科学的表記法が使われます。

```
? 1938000000 * 10000000
1.938E+16 ( 1.938 × 1016 )
```

計算の優先順位

$$? 6 + 2 * 4$$

2つの数式が入った計算の場合は、計算は先頭から行われるでしょうか。

四則演算には、優先順位があります。

優先順位は、演算子表の通りです。

もう一度、例題を実行しましょう。

$$? 6 + 2 * 4$$

$$14$$

かけ算から先に計算されていますね。もし、たし算を先に計算したい時には、カッコを使います。

$$(6 + 2) * 4$$

$$? 32$$

1行の計算式の中に、加減乗除を入れる場合は、先に計算したい式にカッコを付けます。カッコ

は、何重にも付けられますが、数学のように大カッコ、中カッコ、小カッコはありません。いつでも小カッコー種類だけです。

④ ① ③ ②
? 3 * ((8 + 6) / (4 - 2))
2 1

上の例題の計算は、番号の順番に行われます。計算順位が同じであれば、左側の式から計算されます。

カッコを使う場合、左カッコと右カッコが同数でないとエラーになります。必ずカッコの数は確認して下さい。

PRINT 文のもう一つの使い方

PRINT 文の始めに、" " で囲んだ文を使いました。これを計算式に入れると、どうなるでしょうか。

? " 2 + 3 = " ; 2 + 3
2 + 3 = 5

← 必ず入れてください。

今までは、答えだけしか表示されませんでしたでしたが、今後は式も表示されましたね。これは、2つの文の組み合せたものです。

" 2 + 3 " は、計算式でなく、文字として扱われたので、そのまま表示されます。

; (セミコロン) で区切られた後の文は、2 + 3 の式として扱われましたので、答が表示されました。

もう一つの例

? " 2 + 3 = " , 2 + 3

2 + 3 =

5

READY



答が離れた所に出ましたね。(,) コンマを使うと画面の端から 21 桁目の離れた場所に表示されます。

プリント文を使う時は、セミコロンとコンマを上手に使い分けると、見やすい表示が出来ます。

PRINT 文で、(,) コンマや、(;) セミコロンの使い方

```
PRINT "A";"B";"C";"D";"E";"F"
```

```
RUN
```

```
ABCDEF
```

```
PRINT "A","B","C","D","E","F"
```

```
A  _____ 18 字 _____ B  
C                                     D  
E                                     F
```

今まで、色々な計算式を習いましたが、算数で使うイコール (=) 記号が計算式の中に出て来ませんでしたね。コンピュータの場合は、=記号は別の使い方があります。これは、次の章のプログラムの作り方で説明します。

第 3 章 プログラムの作り方

BASIC でプログラムを作ってみましょう。キーボードから入力されたプログラムは、コンピュータ内部のメモリーの中に記憶されます。メモリーには、コンピュータが一連の仕事をする為の順序を示す番号を付けます。

この番号を、行番号（文番号またはラインナンバー）と言います。コンピュータは、行番号の小さい側から、大きい側へと順番に行なわれます。

行番号の後には、文章（ステートメント）を書きコンピュータに仕事（実行）させます。

LET, 変数

例

10	LET	A=	3	CR
20	LET	B=	5	CR
30	LET	C=	A+B	CR
40	PRINT	C		CR
50	END			CR

R U N

C R

8

R E A D Y



RUN という新しい文が出て来ました。これは、コンピュータに仕事を命令する文です。

プログラムを実行させる場合、RUN (ラン) させる、とか、走らせると言います。

LET, 変数

LET (代入文) は、変数に数値又は文字を入れる命令文です。

1 0 L E T A = 3

Aは変数という空箱と考えて下さい。この式は、Aが3に等しいではなく、Aという箱に3という数値を入れる事です。

LET文は、省略する事が出来ますので、

1 0 A = 3


と入力する事が出来ます。

30 LET C=A+B

これも、LET を省略出来ます。

イコール (=) の使い方ですが、等しいという意味ではなく、代入文を意味しています。

変数の図解



A = 5 Aに5を入れる


数値変数



C = A + B CにA+Bの結果を入れる。




=



X = X + 1 Xに1を加えた結果を又同じ変数Xに入れる。

文字変数



代入文の使い方

$X = X + 1$

という使い方があります。算数では有り得ない式ですが、代入文では普通に使われます。

例

```
10  CLS
20  X = X + 1
30  PRINT X ;
40  GOTO 20    ← 命令の実行が行番号 20 へ移ります。
RUN
1  2  3  4  5  6  7  8 . . .
```

1 文字置きに数字が連続して出て来ます。


X は始めは 0 です。

- (0) (0) $X = X + 1$ なので、左辺の X に 1 が代入されます。行番号 40 から行番号 20 に飛び、再び $X =$ ⁽¹⁾
⁽¹⁾ $X + 1$ に戻り X は 1 が加えられて、左辺の X は 2 になります。
この、 $X = X + 1$ の様な使い方はコンピュータでは多く使われます。

文字変数

文字変数の場合は、\$ (ダラー) マークを付けます。

```
10 A=3
20 B=5
30 C=A+B
40 M$="コタエ□" ← スペースを表わします。
50 PRINT M$;C
RUN
コタエ 8
READY
```



```
40 M$="コタエ□" ← スペースキーで1文字空ける。
```

これは、M\$という箱に "コタエ□" という文字のかたまりを入れました。

文字を、一つのかたまりとするには、ダブルクォーテーション、(") で左右を囲んで使います。
もし忘れるとエラーになります。

文字変数の場合は、数字でも、グラフィック記号でも使えます。


また、2つの文字変数を継ぐ事も出来ます。

```

10 A$=" I_"
20 B$=" AM_"
30 C$=" A  BOY "
40 PRINT A$+B$+C$
50 END
RUN
I AM A BOY
READY

```

1文字分あける。



CLS, LIST, NEW

BASICには、色々な命令文があります。プログラムをRUNする場合、画面にプログラムが残っていると邪魔で見にくいものです。

画面を消す命令を、プログラムの先頭に入れておくとRUNをした時に、画面上にある表示を全部消してくれます。

今のプログラムに、新しい文を入れます。

CLS

5 CLS ← 画面の表示を消す命令

今まで入力したプログラムの下の方で、横方向に何も表示がない所に、行番号5 CLSを入力して下さい。

プログラムの入力、行番号1から65535まで使えます。

しかし、1, 2, 3, と行番号を詰めてプログラムを作りますと、すでに入力したプログラムの間に追加が出来ません。ですから、追加が出来る様に、行番号は10番置きにすると追加がしやすくなります。

プログラムを、10番置きに行番号100まで入力している時、行番号25や55など、まだ使っていない行番号を、行番号100の後に入力しても、コンピュータは、内部で並べ替えてくれます。

さき程の、5 CLS がどうなっているか調べてみましょう。

LIST

HOME/CLR **LIST** **CR** と入力します。

```
5  C L S
10 A$ = " I  "
20 B$ = " AM  "
30 C$ = " A  B O Y "
40 P R I N T  A$ + B$ + C$
50 E N D
```

行番号 5 が先頭に入っていますね。
これで実行してください。



LIST は、入力したプログラムを表示させる命令です。

LIST の使い方は、次の方法があります。

LIST	プログラムの内容を全部表示
LIST 行番号	1 行のみ表示
LIST 行番号-行番号	行番号から行番号までを表示
LIST 行番号-	行番号から後ろのプログラムの内容を表示
LIST -行番号	プログラムの先頭から行番号までを表示

LIST で表示されたプログラムは、カーソルを使って、内容を書き替えられます。

```
LIST 30
30 C$="A BOY"
```

カーソルをBの上まで移動して、Mを入力します。続いてAN と入力して **CR** を押して下さい。プログラムの内容を書き替えたら、必ず **CR** キーを押して下さい。

CR キーを押し忘れますと、画面の文字が変わってもメモリー内容は変わりません。

リスト表示中に **SPACE** キーを押しますと、リスト表示が一時中断されます。プログラムをすぐに直したいときは、**BREAK** キーを押して修正してください。リスト表示の中断時に

SPACE キーを押すと、表示が再開されます。

NEW

一つのプログラムが終って、次に新しいプログラムを入力する場合、前のプログラムがメモリー内に残っていると、プログラムが正常に働かない場合があります。
前に入力したプログラムを消すには、CLS文ではメモリー内部から消えません。

プログラムを消すには、

NEW と入力して **CR** を押して下さい。

LISTを出してみましょう。

LIST **CR**

READY



何も表示されませんね。プログラムが、全部メモリーから消えてしまいました。

今までの説明で、少しコンピュータのプログラムが判って来ましたか。

プログラムを入力している時、馴れない内はミスタイプや、文を間違える場合があります。そのままRUNさせますと、ミスのある行番号の所で、実行が止まります。このようなミスを、BUG(バグ)と呼びます。バグとは、虫の事で、虫喰いプログラムと言います。

短いプログラムの場合は、バグを見つけるのは簡単ですが、長いプログラムでは見つけるのが大変です。入力する場合は、落ちついてしましょう。

INPUT. GOTO

計算のプログラムを作ってみましょう。

一番始めに作ったプログラムは、変数の値をプログラムとして置いたので、計算したい数をその都度、修正しなければなりません。これでは、電卓より不便です。

では、連続して使える計算プログラムを作ってみましょう。

```
10  CLS
20  INPUT  A
30  INPUT  B
40  C=A+B
50  PRINT  C
60  GOTO  20  ← 飛び先の行番号
RUN  CR
```

? (INPUT Aの入力待ち) 4 CR — 4を入力

? (INPUT Bの入力待ち) 5 CR — 5を入力

9

? 8 CR

? 7 CR

15

? ここで BREAK キーを押す

BREAK IN 20 — 行番号20で実行を

READY 中止しました。


GOTO

プログラムの流れが、GOTO 文に出会うと指定された行番号に、無条件で飛びます。

このプログラムは、行番号 60 から行番号 20 に戻り、再び、INPUT A から始まります。このようなプログラムは、無限にグルグル廻りますので無限ループと呼ばれ終わりがありません。プログラムを止めるには、**BREAK** キーで止めるしかありません。

プログラムの流れが、INPUT 文に出会うと、キーボードから変数に数値が入力され、**CR** キーが押されるまでプログラムの流れが止まっています。

INPUT 文は、文字変数も使えます。また、PRINT 文と同じに、" " で囲んだ文字も付けられます。

```
NEW
10 CLS
20 INPUT "ナマエハ?_";A$
30 PRINT A$
40 END
RUN
ナマエハ? ハナコ ←文字を入力
ハナコ
READY

```

END, STOP

END はプログラムの終りを知らせます。

STOP は、プログラムの流れを止めます。

CONT

STOP 文や、BREAK キーで中断されたプログラムを続けて実行したいときに使います。

```
10 X=X+1  
20 PRINT X  
30 GOTO 10
```

このプログラムをRUNさせ、途中で **BREAK** キーで実行を止めて下さい。

```
Break in 20  
CONT CR
```

これで再びプログラムが止まったところから再開されます。

FOR ~ TO, NEXT, STEP

一定回数、繰り返して仕事をさせる時に使います。

```
10 CLS
20 FOR N=0 TO 9
30 PRINT N
40 NEXT N
50 END
```

← 10回くり返します。

FOR ~ TO は、NEXT と一対で使います。このプログラムは、N が0 から9まで1つずつ増えます。

```

10 CLS
20 FOR N=0 TO 20 STEP 2
30 PRINT N;
40 NEXT N
50 END
RUN
0 2 4 6 8 10 12 14 16 18 20

```

STEP 2 は、きざみ幅を2にしましたので、0 ~ 20 を2つつ増しています。

STEP は、-（マイナス）も使えます。行番号 20 を修正してみましょう。

```

20 FOR N=20 TO 0 STEP-2
RUN
20 18 16 14 12 10 8 6 4 2 0

```


20 から、2つつつ少なくなりましたね。このように一定数、増減させる場合に使います。

FOR ~ NEXT 文は、重ねて使えます。

```

10 CLS
20 FOR A=1 TO 9
30 FOR B=1 TO 9
40 PRINT A*B;
50 NEXT B
60 PRINT
70 NEXT A

```




FOR～NEXT文を重ねるのを、"入れ子"といいます。4重まで"入れ子"に出来ます。
それ以上重ねると、ネスティング・エラーになります。

FOR I=1 TO N のように、変数も使えます。

```

10 CLS
20 FOR N=1 TO 20
30 FOR M=1 TO N
40 PRINT "●";
50 NEXT M
60 PRINT
70 NEXT N

```



正しい使い方

入れ子にする場合の注意

```
FOR N=1 TO 20  
  FORM=1 TO 10 .....  
NEXT N  
NEXT M
```



正
し
く
な
い
使
い
方

FOR, NEXT のグループを交差させることは出来ません。

IF ~ THEN. GOSUB

プログラムの流れは、行番号の小さい側から順番に進みますが、ある条件になった場合に、流れを変えるようにしてみましょう。

【合格 不合格の適性プログラム】

```
10 CLS  
20 INPUT "トクテン" ; A  
30 IF A >= 65 THEN GOSUB 100  
40 IF A < 65 THEN GOSUB 200
```

```

50 GOTO 20
100 PRINT "ゴウカク"
110 RETURN
200 PRINT "フゴウカク"
210 RETURN

```

IF～THEN は、条件判断をする命令です。上のプログラムは、INPUT 文で入力された得点を、IF 文で判断して流れを変えております。

入力された得点が、65 点以上の場合は、GOSUB 文で行番号 100 に行き、65 点に満たない場合は、行番号 200 に行きます。

(もし) (ならば) (行ってから戻れ) (行番号)
 IF A>=65 THEN GOSUB 100

IF～THEN の後には、行番号以外の文も使えます。

(省略出来る)

IF～THEN GOTO 行番号 (指定行番号にとびます。)

IF～THEN GOSUB 行番号 (指定行番号にとびます。)

IF～THEN PRINT "xxx" 画面に出力します。

IF～THEN END (プログラム終了)

IF～THEN STOP (プログラム実行中止)

IF～THEN BEEP (音を出します)

条件判断は表の比較演算子を使って下さい。

比較演算子表


	記号	内 容
比較演算	=	等しい (True なら - 1 , False なら 0)
	<>	等くない (True なら - 1 , False なら 0)
	>	大きい (True なら - 1 , False なら 0)
	<	小さい (True なら - 1 , False なら 0)
	>=	以 上 (True なら - 1 , False なら 0)
	<=	以 下 (True なら - 1 , False なら 0)
論理演算	NOT	論理反転
	AND	論理積
	OR	論理和
	XOR	排他的論理和

GOSUB, RETURN

前のプログラムに、GOSUB文がありました。

GOTO 文の場合は、指定された行番号に行くだけでしたが、GOSUB 文は RETURN 文と一語に使われ、指定された行番号に飛んでから、RETURN 文で GOSUB 文の次の行に戻って来ます。


```
20 INPUT A
30 IF A >= 65 THEN GOSUB 100
40 IF A < 65 THEN GOSUB 200
50 GOTO 20
100 PRINT "ゴウカク"
110 RETURN
200 PRINT "フゴウカク"
210 RETURN
```



GOSUB 文を使う時に、RETURN 文を忘れると結果がおかしくなります。

RETURN 文で戻った後、GOSUB の次の行からプログラムは進行しますので、行番号 50 のように、もう一度、プログラムの流れを変えて下さい。

プログラムが、途中から分岐する場合、流れがよく判るように、フローチャートを作ります。
これは、流れ図とも言われます。



フローチャートは、上から下に進みます。

条件判断文で、流れが変わります。複雑なプログラムを作る時は、フローチャートを作ると、流れがはっきり判ります。

ON GOTO

条件文に似た使い方で、ON GOTOがあります。

```
ON A GOTO 100, 200, 300
```

変数Aが、1の時、行番号100に飛びます。


変数Aが、2の時、行番号200に飛びます。

GOTO文の後に来る、行番号の数だけ変数の値が使えます。

下のプログラムでは、Aは1～3までの値を入力してください。

```
10 INPUT "ゴチューモンハ" ; A
20 ON A GOTO 100, 200, 300
100 PRINT "コーヒー" : GOTO 10
200 PRINT "ケーキ" : GOTO 10
300 PRINT "ミルク" : GOTO 10
                        ↓(コロン)
```

1行の中に、2つ以上の命令文を入れる事が出来ます。(マルチステートメント)
独立した文を一行の中に2つ以上入れる場合は、:(コロン)で区切って下さい。




ON GOSUB も、同じ使い方をします。説明の前に新しい命令を使ってみます。

CURSOR

今まで、RUN した時、表示が左側に出るだけでした。これを、画面上のどの位置に出すか、プログラム上で指定した方が見やすくなります。

CURSOR 文は、表示位置をきめる命令です。




プログラムを書き込む画面（テキスト面）は、38桁×24の912のマス目で構成されています。

CURSOR 18,12 : PRINT "A"

行番号を使わずに、直接入力して下さい。

Aの文字が画面の中央に表示されましたね。

グラフィック画面で CURSOR 文を使う場合は、



x 軸方向 0 ~ 255 (256 ドット)

y 軸方向 0 ~ 191 (192 ドット)

CURSOR 文で座標を指定しますと、表示したい文字の先頭位置がきまります。

ON GOSUB

```
10 CLS
20 CURSOR 10,3:PRINT"メニュー"
30 CURSOR 10,6:PRINT"1...ノミモノ"
40 CURSOR 10,8:PRINT"2...タベモノ"
50 CURSOR 10,10:PRINT"3...デザート"
60 CURSOR 10,13:INPUT"ナニニシマスカ";A
70 ON A GOSUB 100,200,300
80 GOTO 60
```

前に表示したものを消す。

```
100 CURSOR 10,16:PRINT" "
110 CURSOR 10,16:PRINT"コーヒー...¥300"
120 RETURN
200 CURSOR 10,16:PRINT" "
210 CURSOR 10,16:PRINT"ケーキ... ¥200"
220 RETURN
300 CURSOR 10,16:PRINT" "
310 CURSOR 10,16:PRINT"メロン... ¥250"
320 RETURN
```

GOTO の時のプログラムを、変型したものです。CURSOR 文ばかりですが、表示位置が中央に来ているので、見やすくなります。

GOSUB を使いますので、RETURN は忘れずに入れて下さい。

READ, DATA, RESTORE

プログラムの中に、DATA を入れておき、READ 文で読み出す事が出来ます。

```
10 READ A, B, C, D
20 PRINT A+B+C+D
100 DATA 1, 2, 3, 4
RUN
10
READY
```




DATA の数値を足し算して結果を出しました。

DATA は、文字も扱えます。

プログラムの流れは、READ 文に出あうと、DATA 文がどこにあっても、先に読み取ります。

```
10 READ A$,B$,C$,D$
20 PRINT A$+B$+C$+D$
30 DATA S,E
40 PRINT
50 DATA G
60 DATA A
RUN
SEGA
```



1行分のスペースを空ける

READY

文字変数を使った場合、DATAに数字を入れても、文字として扱われて、算数計算には使えません。

DATAの数と、READ文の変数の数は、同数でなければなりません。

DATA数が、多過ぎるとREAD文の変数分だけしか、DATAは出て来ません。

READ文の変数が、DATAより多いとエラーになります。

同じDATAを始めから繰り返して使う場合は、RESTORE文を使います。

RESTORE

```
10 READ A, B, C, D
20 DATA 1, 2, 3, 4
30 RESTORE
40 READ E    ← 始めのデータ、1を読む
50 PRINT A+B+C+D+E
RUN
11
```

DIM (配列)

一次元配列

前のプログラムでは、DATA に対して、A, B, C, D と変数を使いました。データ数がふえると、変数を一つずつ設定するのは大変な事です。

この様な場合、配列を使います。

DIM A(5) ← カッコの中の数値を添字と言います。

この意味は、

A(0), A(1), A(2), A(3), A(4), A(5)

6 変数の変数を宣言した事になります。

文字変数も配列宣言が出来ます。

```
10 CLS
20 DIM A$(5), B(5)
30 FOR I=0 TO 5
40 READ A$(I), B(I)
50 PRINT A$(I), B(I)
60 PRINT ← 一行、間を空けるため、
70 NEXT I
100 DATA コーヒー, 300, ミルク, 150, ケーキ, 200
110 DATA ティー, 280, トースト, 180, パン, 100
```

DIMA\$(5)としたのに、READ文でA\$(I)になっているのは、FOR I = 0 TO 5で使っている、Iです。ですから、Iが0から5に変化しながら、DATAを読み込んでいます。

READ文では、A\$とBを続けて読み込みますから、DATAは、文字変数と、数値の変数を交互に並べます。

DATAの並べ方は、実験してみてください。

二次元配列

二次元配列は、DIMA(5, 5)のように、カッコの中の添字が2つに分れます。

九九の表

```
10  CLS
20  DIM A(9,9)
30  FOR J=1 TO 9
40  FOR K=1 TO 9
50  A(J,K)=J*K
60  PRINT A(J,K);
70  NEXT K
80  PRINT ← 行を変えるため
90  NEXT J
RUN
```

1	2	3	4	5	6	7	8	9
2	4	6	8	10	12	14	16	18
3	6	9	12	15	18	21	24	27
4	8	12	16	20	24	28	32	36
5	10	15	20	25	30	35	40	45
6	12	18	24	30	36	42	48	54
7	14	21	28	35	42	49	56	63
8	16	24	32	40	48	56	64	72
9	18	27	36	45	54	63	72	81

三次元配列

```
DIM A(5,5,5)
```

配列宣言は、3次元までです。

DIM文で配列宣言しない場合は、自動的に添字の最大値は10です。

ERASE

プログラムの途中で、配列宣言を無効にしたい時に使います。

100 ERASE としますとプログラム内部の配列は、全部無効になります。

100 ERASE A, B\$ のように、配列名を入れますと、その配列を無効にします。

DELETE (デリート)

プログラムを修正する場合、不要な行を消す時に使います。

DELETE 180-220 CR ; (コンマ)でも良い、行番号180-220 までを取り消します。

DELETE -250 先頭から、行番号250までのプログラムを消します。

DELETE 600- 行番号600から後を全部消します。

DELETE 100 行番号100だけ消す。

今まで、BASICの文の一部を使って来ました。

プログラムを作る場合、必らず行番号を入れます。行番号を自動的に作り出す文があります。

AUTO (オート)

AUTO C R と、行番号なしで入力して下さい。

10 C R

20

行番号が、10行きざみで自動的に発生して来ます。

AUTO 100 C R

100 C R

110

行番号が、100番から、10きざみで出て来ましたね。

AUTO, 10, 20 C R

(始めの行番号、きざみ幅)

10 C R

30 C R

50

C R キーを押すたびに、20きざみで行番号が出て来ます。

RENUM (リナンバー)

プログラムを入力していて、行番号を追加している内に、詰め過ぎた場合に、行番号を付け直したい場合に使います。

RENUM CR

プログラムの先頭から、10, 20, 30 と行番号を付け直します。

RENUM 100 CR

行番号100から、10きざみに直します。

RENUM 300, 200

 ↓ └─旧番号
 新番号

行番号200からのプログラムを、行番号300から10きざみにします。

RENUM 300, 200, 50


 └きざみ幅指定

行番号300から50きざみにします。

SAVE, LOAD, VERIFY

SAVE

作成したプログラム、データ等をカセットテープに記録して置く事が出来ます。



プログラムの内容が判る名前を付けて下さい。(ファイルネーム)

名前は、16字以内で付けて下さい。

カセットデッキは、お手持ちのものをお使い下さい。

カセットデッキに、カウンターが付いていましたら、カウンターの数字をカセットテープのラベルに記入しておいて下さい。

VERIFY

SAVE が正確に行われたかをチェックします。

テープを巻き戻し、

VERIFY **CR** と入力して、プレイボタンを押して下さい。

正しく、SAVE されていれば、Verify end と表示されます。

end が出ない場合は、RESET キーで止めて、始めから SAVE 仕直して下さい。

LOAD

カセットテープに入っているプログラムデータをコンピュータに移します。

LOAD " x x x " **CR** プログラム名

Loading Start ← プレイボタンを押す。

Found " x x x "

Loading End ← カセットを止めて下さい。

SAVE, LOAD 等は、セガ製のデータレコードをお使い下さい。

もし、お手持ちのカセットデッキを使う場合、音のレベルにより、カセットテープに書き込みや読み取りが出来ない場合があります。

これは、コンピュータの故障ではなく、カセットデッキの性能による場合があります。

音量や音質のレベルを変えて試して下さい。それでも書き込みが出来ない場合、セガ製データレコードを使用して下さい。

カセットテープレコーダとの接続は、セガ製ミニプラグを使って下さい。

市販品をお買いになる場合は、抵抗の入っていないものを指定して下さい。

SC-3000 側

カセットレコーダ側

IN ←	LOAD, VERIFY	→ イヤホン (EAR)
OUT ←	SAVE	→ マイク (MIC)

REM

プログラムを作成する時にプログラムの内容や、プログラム中のサブルーチンの内容が判るように、注釈文を入れておくと後でプログラムリストを見たとき便利です。


```
10 REM   ×××   ケイサン   ×××
20 CLS
30 PRINT 2+3
}
```

REM文は、プログラム中では無視されて実行はされません。

CONSOLE

テキスト画面のスクロール範囲、クリック音の ON, OFF, 英文字の大小文字の切り替えを指定します。

スクロール範囲指定



V L C S
CONSOLE 5, 15, 0, 1

V; スクロールの上限 (0~22)

L; " 長さ (2~24)

C; クリック音の有無 0:無し
1:有り

S; 英数の大小 0:シフトなし大文字
1:シフトなし小文字

画面は、y 方向に 0~23行に割られております。CONSOLEの後の数字は、スクロール開始行、スクロール終了行を示します。

例のように設定しますと、5行目から15行目までの間だけカーソルが移動します。

カーソル移動範囲の設定は、0から23までにしてください。24以上を設定しますと、エラーになり

ます。

スクロールの長さは、2 から 2 4（上限の設定数）まで使えます。

CONSOLE ,, 0 3番目の数字は、キーを押した時のクリック音を ON, OFF します。

0 ; 音が出ない。

1 ; 音が出る。

CONSOLE ,,, 1 4番目の数字は、英字の大小を決めます。

0 ; シフトなしで大文字

1 ; シフトなしで小文字

4 つの数字は、省略する場合（ , , , ）だけを入力して下さい。

CONSOLE 文は、RESET キーで解除出来ます。

第 4 章 関 数

関数のグループには、数学的関数と文字関数があります。良く使われる関数を覚えましょう。

R N D (ランダム)

変数の値をランダムに発生させます。この利用方法は、サイコロの目を出したり、ゲームで標的を不規則に動かしたりするのに便利なのでよく使われます。

R N D (1) 続けて乱数が発生する。

R N D (0) 乱数系列が初期化する。

R N D (- 1) 乱数系列を入れ変える。

乱数を出してみましょう。

```
10  FOR  N=0  TO  10
20  R=RND( 1 )
30  PRINT  R
40  NEXT  N
RUN
```

. 2 3 8 0 5 4 6 2 9 4

. 7 0 4 1 3 8 2 4 9 6

. 4 9 2 5 3 7 1 1 3 8

}

0から1未満までの間の数字がデタラメに出て来ましたね。このデタラメが乱数なのです。次に出て来る数字が予測出来たのでは、乱数といえません。サイコロの目も振ってみなければ何が出るか、わからないから面白いのです。

でも小数以下の数字では実用的ではありません。これをもっと使いやすくします。

INT (n) インテジャー

INT関数は、小数点のある実数を整数に変えます。

? INT (3.1 4)

CR

3 ← 小数が消える。

READY

サイコロ

10 FOR N=0 TO 20

20 S=INT (RND (1) * 6)

30 PRINT S ;

```

40 NEXT N
RUN
3 5 0 4 1 5 0 .....

```

小数点がなくなりましたが、0があつて6がありませんから直しましょう。

```

20 S=INT(RND(1)*6)+1

```

└─ 0を出さない為に必要な数までにする
ため

└─ 出したい数

こんどは、1から6までの数が出ましたね。数値を色々変えて試して下さい。

四捨五入プログラム

```

10 INPUT A ← 小数付き数値
20 PRINT INT(A+0.5)
30 GOTO 10

```

小数点以下を四捨五入するプログラムです。これは、いろいろな計算に応用が出来ますので覚えましょう。

DEF FN

コンピュータを使う人が自由に定義する関数です。

```
5  REM エンノメンセキ
10  DEF  FNS ( R ) = R * R * 3 . 1 4 1 5  ← 円周率
20  INPUT " ハンケイ = " ; A
30  Z = FNS ( A )
40  PRINT
50  PRINT " メンセキ = " ; Z
60  END
RUN
```

ハンケイ = 1 0

メンセキ = 3 1 4 . 1 5

DEF FNS (R) = R * R * 3 . 1 4 1 5 は、右辺の式を左辺の FNS (R) に関数として定義しておきます。

半径を入力しますと、行番号 30 で定義した関数を呼び出して計算します。

(R) を仮引き数といい、この中に入るデータを実引き数といいます。

文字列関数

前におぼえた文字変数についてももう少し詳しく説明しましょう。

文字変数は、文字列変数ともいいます。他のテキストでは、ストリング変数と書いたのがあります。

コンピュータの中では、文字はどのようにあつかわれているか調べてみましょう。

ASC ("n") アスキー関数

ASCII (アスキー) とは、American Standard Code Information Interchange (アメリカンスタンダードコードインフォメーションインターチェンジ) アメリカの基準コードで文字や記号に番号をつけて、コンピュータが情報を処理しやすくされています。

では試してみましょう。

?ASC("A") CR ダイレクトモードで実行

65

Ready

この65という数字は、Aを現わしています。ASC文を使うときは("A")のように、カッコ内の文字を" "で囲んで下さい。

Aの替りに、今度は記号を入れてみて下さい。

?ASC(" ! ") 今度は33と出ましたね。

コンピュータの内部には、キーボードにある文字や記号が32番から255番までの数字に対応して入っています。

コンピュータは、人間の様に文字をそのまま判断出来ません。すべて数字として扱います。ですから、A(65)はB(66)より先にあると判断して文字を分類することが出来ます。

?ASC("BA")のように、2つ以上入れても始めの一文字だけ調べて数字を出します。

もう一つ試みましょう。

```
10 INPUT A$
20 Q=ASC(A$)
30 PRINT Q
40 GOTO 10
```

RUNしたあと、文字や記号を入力するとそれに対応したコード番号が出て来ます。

CHR\$

ASC文と対になっていて、数値を文字やコントロール機能を与えます。

```
?CHR$(65) CR
A
```

Aは、ASCII コードでは65でしたね。

コンピュータの中に入っている文字をのぞいてみましょう。

```
10 FOR M=32 TO 255
20 PRINT CHR$(M);
30 NEXT M
RUN
```

キーボードに印字されている文字や記号が並んで出て来ましたね。これが、コンピュータの内部にある文字や記号です。

キャラクターセットを見て下さい。コード表と画面上に表示されたコードが同じですね。

LEFT\$, RIGHT\$, MID\$

長い文字列の中から、文字の一部を取り出す関数です。

```
10 A$=" コーヒー ココア ミルク "
20 M$=LEFT$(A$,4)
30 PRINT M$      · ↑ (左から4番目までの文字)
RUN
コーヒー
```

A\$の中にある文字列(スペースも数えます)の4番目までを取り出して、M\$の中に入れ画面に表示させます。

```
10 A$="コーヒー ココア ミルク"  
20 M$=RIGHT$(A$,3)  
30 PRINT M$      ↑ 右から3番目までの文字  
RUN  
ミルク
```

こんどは、右から3番目を起点として終りまでの文字を取り出します。

```
10 A$="コーヒー ココア ミルク"  
20 M$=MID$(A$,6,3)  
30 PRINT M$      ↑  ↑ 取り出す文字数  
RUN              起  
ココア          点
```

文字列の左から6番の文字を起点として3つの文字を取り出します。

LEN (レングス)

LEN(A\$)は、A\$の文字数を教えてくれます。この場合も、" "で囲まれた文字全部で、

スペースも含まれます。もし " " の間が全部スペースでも文字としてあつかいます。

```
10 A$ = " SEGA PERSONAL COMPUTER "  
20 PRINT LEN(A$)  
RUN  
22
```

文字の数をスペースを含んで教えてくれました。

こんな使い方もあります。

```
10 A$ = " ***** "  
20 FOR I = 1 TO LEN(A$)  
30 PRINT LEFT$(A$, I)  
40 NEXT I  
RUN  
*  
**  
***  
****  
}  
*****
```

STR\$, VAL

数値を文字変数に変換したり、文字変数として使われている数字の文字列を数値に変換します。

STR\$

```
10 A=1:B=3
20 D$=STR$(A)+STR$(B)
30 D=A+B
40 PRINT D$,D
RUN
1 3          4 ← 行番号30の結果
└─ 行番号20の結果
```

STR\$(A)としますと、数字が文字に変わります。文字の足し算は、文字が並ぶだけで計算としての答は出て来ません。

VAL

VAL関数は、STR\$とまったく正反対の働きを持ち、文字列の数字を数値に直します。

```
10 A$="12345"
20 B$="11111"
```

```

30 C$=A$+B$      文字列のたし算
40 C=VAL(A$)+VAL(B$) ← 数値のたし算
50 PRINT C$
60 PRINT C
RUN
1234511111 ← 文字列
 23456      ← 数値

```

TIME\$

コンピュータの内部には、時計機能があります。水晶発振の正確なクォーツ・デジタル時計です。コンピュータの電源を ON しますとその時から1秒きざみで働き始めます。

電源を入れた時 00:00:00 です。

一定時間が過ぎると、その時間だけ表示します。

```

PRINT TIME$ [CR]
00:12:32 ← 電源を入れてからの時間

```

時計として使う場合は、

```
10 TIME$ = "08:15:00" ←現在の時刻 ("時:分:秒")
20 CURSOR 15,15:PRINT TIME$
30 GOTO 20
```

一度時刻を入力しますと、Reset ボタンを押すか、電源を切るまで残っています。

これでデジタル時計になりました。

SPC (スペース), TAB (タブレーション)

PRINT文で使います。

SPC関数は、文字と文字の間のスペースを指定します。

```
10 PRINT "ABC";SPC(10);"XYZ"
RUN
ABC      XYZ
```

10ケの空白(スペース)

SPCで指定した範囲に文字があると消されてしまいます。

TAB関数は、画面の端から何文字目に出すか指定します。

```
10 PRINT TAB(5);"ABC"
RUN
```

□□□□□ A B C

5文字分のスペース

T A B 関数の場合は、指定した場所までの間に文字があっても消しません。
この関数は、P R I N T 文の時に使いますのでよく覚えましょう。

I N K E Y \$

文字や数字のどのキーが押されたかを見る文です。

ゲームのようにキーボードで何かの絵（キャラクター）を動かすのに便利です。

```
10 X$=INKEY$  
20 IF X$=" " THEN 10  
30 PRINT X$;  
40 GOTO 10
```

行番号20でキーが押されているかを見ています。何も入力されていないときはX\$の値はスル（何もしない事）ストリングと言って、何も表示もせず行番号10と20の間をぐるぐる廻っています。（無限ループ）

もし何かのキーが押されますと、そのキーの値がX\$の中に入り行番号30で表示されます。この無限ループから抜けるには、

```

25 IF X$ = "Z" THEN 100
100 PRINT "END" : END

```

と追加して下さい。

これでZキーを押すとプログラムの実行が終了します。

例 , キーで、動きます。

```

10 DIM D (29)
20 CLS
30 X=18: Y=20
40 D (29) = -1 : D (28) = 1 : D (0) = 0
50 K$ = INKEY$
60 IF K$ = " " THEN K = 0 : GOTO 90
70 K = ASC (K$)
80 IF K > 29 THEN K = 0
90 X = X + D (K)
100 IF X < 0 THEN X = 0
110 IF X > 33 THEN X = 33
120 CURSOR X , Y : PRINT " | + | "
130 GOTO 50

```

FRE

プログラムを入力しますと、残りのメモリーが減っていきます。あと、どれぐらいメモリーが残っているか知りたいときに、FRE関数で調べます。

例

```
PRINT FRE  
8300
```

あと8300BYTE(バイト)のプログラムを入れることができます。

プリンター制御命令

LLIST

プリンターにプログラムリストを書きます。

○命令文の使い方は、LISTと同じです。

LLIST		プログラム全部を書きます。
LLIST	行番号	指定した行番号を書きます。
LLIST	行番号 - 行番号	
LLIST	- 行番号	
LLIST	行番号 -	

OUT

出力ポートにデータを出力させる命令です。

○出力ポート番号とは、外部に対しデータを送り出すため、システムの中であらかじめ決定されています。

出力ポート番号は、0～255（&H00～&HFF）までの整数です。

VDP	データレジスタ	&H	BE
	コマンドレジスタ	&H	BF
	サウンドジェネレータ	&H	7F

POKE, PEEK, CALL

BASICでプログラムを入力すると、順番にメモリーに記憶されます。それ以外に、特定のメモリーに、データや機械語を書き込むことが出来ます。

POKE (ポーク)	アドレス	データ
書き込み命令	POKE	&H9000, 65

アドレスは、&H8000（-32768）～&HFFFFまでです。

データは、0～255までの整数です。

尚アドレスは、メモリー使用量や、BASICのバージョン、種類によって異なります。

PEEK (ピーク) 関数

アドレス

読み出し命令

A = PEEK (&H9000)

指定されたアドレスのメモリーの内容を読み出します。

メイン・メモリ マップ

&H0000

⌋

&H7FFF

&H8000

⌋

&HFFFF

BASIC

領域

テキスト領域

ROM (読み出し専用)

POKE 命令は使えません。

RAM 領域

(32 KBYTE 実装時)

DATA ヘンカン プログラム

```
10 REM *** DATA ヘンカン
20 INPUT "DATA=" ; D
30 IF D=>256 THEN GOTO 20
40 POKE&H9000,D
50 A=PEEK(&H9000)
70 B$=CHR$(A)
80 PRINT A;"=";B$
90 GOTO 20
RUN
DATA=65
65=A
DATA=■
```

入力した数値を記号に変えるプログラムです。数値が256以上の場合は、もう一度入力し直しになります。

記号が出てこない場合は、その数値に対応した記号が無いからです。

CALL

機械語で書き込んだアドレスを呼び出します。

機械語（マシン語）は、**BASIC**言語と違い別に勉強をしなければなりません。

間違った使い方をすると暴走してプログラムを壊してしまいますので注意しましょう。

機械語の勉強は、別な機会にしましょう。

LPRINT

プリンターにPRINT文の内容を書きます。

○命令文の使い方は、PRINT文と同じです。

LPRINT A ← Aの内容をプリンターに書きます。

10 INPUT A, B

20 C=A+B

30 LPRINT C

40 GOTO 10

RUN

プリンターにCの値が書き出されます。

HCOPY

TV画面に表示されている、文字や記号がプリンターに書き出されます。

しかし、プリンターに書けるのは、数字、英字の大小文字、カナ、ASCIIコードの記号のみです。グラフィックモードの記号は書けません。

第 5 章 グラフィック

グラフィックを使いましょう。

いよいよ、カラーを使って図形を描いてみましょう。色々な図形や、色を塗ったりするのは案外簡単です。エラーを恐れずに試してみましょう。

COLOR

文字や、キャラクター、下地の色を指定します。

例

COLOR 1, 5, (100, 100) - (150, 150), 7

ビット " 1 " の色 バックドロップの色
ビット " 0 " の色 背景色をセットする範囲

テキスト画面（プログラムを入力する画面）

ビット " 1 " の色は文字などの色を指定します。

例では、文字が黒、地は青になります。

グラフィック画面（グラフィック文で表示する画面）

ビット " 1 " の色は、

PRINT文による文字の色

LINE, PSET 文による線や点の色。

LINE,

PAINT文による塗りつぶしの色。

COLORを使った後は、
RESETキーを押すと
元の状態に戻ります。

ビット"0"の色

下地の色

BLINE, PRESET, によりビット"1"が消された時の色。

例

```
10 CLS                                40 FOR I=0 TO 300
20 FOR C=0 TO 15                    50 NEXT I,C
30 COLOR 1,C
```

画面の色が次々と変わります。

透明色とは、そのままバックドロップの色がでます。

色番号	色	色番号	色	色番号	色
0	透 明	6	こ い 赤	12	こ い 緑
1	黒	7	水色(シアン)	13	マゼンダ
2	緑	8	赤	14	灰
3	うすい緑	9	うすい赤	15	白
4	こ い 青	10	こ い 黄		
5	うすい青	11	うすい黄		

SCREEN

書き込み画面、表示画面を選びます。

テキスト画面しか使わないのであれば、SCREEN指定はいりません。
グラフィック面に文字や絵を表示するときに使います。


SCREEN 書き込み画面、表示画面
 └───┘
 *

* ┌ { 1 : テキスト画面
 └ { 2 : グラフィック画面

書き込み画面の指示


表示画面の指示

PRINT又は
CLS等の書き
込み



LINE

線を引きます。(SCREEN 2,2を入力してから使います)



グラフィック画面は、 x 方向は0から255 (256ドット) y 方向は0から191 (192ドット) の座標を持っています。

LINE文は、2点間の座標を指定して線を引きます。

画面の例


(1) LINE (50 , 50) - (200 , 50) , 1 黒色指定

(2) LINE (50 , 100) - (200 , 150) , 8 赤色指定

LINE文を連続して使う場合、書き始めの座標を省略しますと以前に描いた座標から線を引きます。

```
10 SCREEN 2, 2:CLS
20 LINE (50, 50)-(150, 50), 1
30 LINE-(50, 150), 8
```



線を描き終ると、すぐにテキスト画面に変わります。SHIFTキーを押して  キーを押して下さい。画面が変わりグラフィック面に線が描かれてるのがわかります。

箱を描く (BOX)

LINE文で対角線を指定して、色の後にBを付けて下さい。

```
10 SCREEN 2, 2:CLS
20 FOR C=0 TO 15
30 LINE (80, 50)-(160, 100), C, B
40 FOR I=0 TO 300:NEXT I
```

```
50 NEXT C
```

```
60 GOTO 60 (無限ループにしました。BREAKキーで止まります)
```

箱の中を塗りましょう。行番号30のBの後にFを入れて下さい。Cの色指定できれいに塗り変えられていきます。

BLINE

LINE文で描いた線や、箱を消します。使い方は、LINE文と同じですが色指定は無視されますのでいりません。

```
10 SCREEN 2, 2:CLS
```

```
20 LINE (50, 50)-(200, 50), 1
```

```
30 FOR I=0 TO 300:NEXT I ← 時間をとる。
```

```
40 BLINE (50, 50)-(200, 50)
```


```
50 GOTO 50
```

箱を消す時も同じですが、描いた箱より小さい箱をBLINEで描きますと、その部分だけ色が消えます。

BLINE文は、一度描いた絵を全部消したり、一部分だけ消したりするのに使います。


PAINT

画面に LINE 文や CIRCLE 文で区切られた部分を塗ります。



PAINT (x, y), 色

塗り始めの座標線で囲まれている部分を塗ります。



まわりを全部塗ります。

LINE , BF で描いた箱

線で区切る場合、すき間を作らないようにして下さい。


PSET

画面上の指定された位置に点を打ちます。

座標 色

PSET (x , y) , 1

座標を連続的に変化させると直線や曲線が描けます。



点が描ける

```
10 SCREEN 2,2:CLS
20 X=0:Y=95:E=1
30 PSET(X,Y),8
40 X=X+1:Y=Y+E
50 IF Y=120 THEN E=-1
55 IF Y=85 THEN E=1
60 IF X=250 THEN END
70 GOTO 30
```

PSETは、点をちりばめたり関数を使ったグラフを作ったりします。

PRESET

PSETと反対に点で消します。使い方はPSETと同じで点を描く替りに消す役目です。使い方は同じです。


座標

PRESET(x, y)

POSITION

画面の座標は左上が $x = 0, y = 0$ です。

POSITION文で座標を指定すると、その位置が中心点、 $x = 0, y = 0$ になります。



x 軸 y 軸

POSITION(100, 80) 0, 0

1 1— y の軸方向

└— x の軸方向


座標の後の数値は、 x 軸 y 軸の増加方向をきめます。

0, 指定 x は右に増加
 y は下方に増加します。

1, 指定 x は左に増加
 y は上に増加します。


x , y の指定を組み合わせると x 軸,
 y 軸の増加分を変化させられます。

POSITION(100, 80)



```

10 SCREEN 2, 2:CLS
20 POSITION(125, 95), 1, 1
30 FOR N=0 TO 50
40 PSET(X, Y), 1
50 X=X+1:Y=Y+1
60 NEXT N
    
```



POSITION 文は、PSET と一諸に使うと関数グラフなどが描けます。

```
10 SCREEN 2,2:CLS
20 POSITION(100,50),0,0
30 FOR N=-10 TO 1 STEP .1
40 X=N*20+120:Y=SIN(N)*50+45
50 PSET (X,Y),1
60 NEXT N
```

CIRCLE

こんどは、円を描いてみましょう。線画や円内に色を付ける事が出来ます。

CIRCLE文には、色々な数値が入りますから覚えるまでは、テキストを見ながら入力して下さい。




	(1)	(2)	(3)	(4)	(5)	(6)	(7)
	座標	半径	色	比率	始点	終点	
CIRCLE	(125 , 95) ,	50 ,	5 ,	1 ,	0 ,	1 ,	BF

例の説明

- | | | |
|--------|-----------------------|--------------------------------|
| (1) 座標 | $x = 125$, $y = 95$ | (x 方向最大 255 , y 方向最大 191) |
| (2) 半径 | 中心点からの半径 | |
| (3) 色 | 0 ~ 15 まで | |
| (4) 比率 | 1 のとき | 真円 |
| | 1 より小さい | よこ長のだ円 |
| | 1 より大きい | たて長のだ円 |
| (5) 始点 | 書き始めの位置 | |
| (6) 終点 | 書き終りの位置 | (0 から 1 までの間で少数点で入力する。) |
| (7) | BF を指定しないときは円周のみ描きます。 | |

Bを指定すると内側にも線が引かれます。

BFにすると(3)で指定した色で塗りつぶします。



```

10 SCREEN 2,2:CLS
20 FOR R=10 TO 50 STEP 10 (半径が10ずつ増えます)
30 CIRCLE (125,95),R,8,1,0,1
40 NEXT R
50 GOTO 50

```

BCIRCLE

CIRCLE 文で描いた円を消す時などに使います。この場合は色指定は無視され、下地と同じ色で円を描きますので見えなくなります。

前のプログラムに追加してみましょう。

```

50 FOR R=10 TO 50 STEP 10
60 BCIRCLE(125,95),R,1,1,0,1
70 NEXT R
80 GOTO 10

```

PATTERN

PATTERN 文を使って文字や絵（キャラクター）を作る事が出来ます。

テキストモードの文字を書き変えてみましょう。

PATTERN C# 文字コード (32 ~ 255 又は、&H20~&HFF) , "文字列式"

(文字列式は16進数)

左	右	左	右	左	右
		0 1 1 1	0 0 0 0	7	0
		1 0 0 0	1 0 0 0	8	8
		1 0 0 1	1 0 0 0	9	8
		1 0 1 0	1 0 0 0	A	8
		1 1 0 0	1 0 0 0	C	8
		1 0 0 0	1 0 0 0	8	8
		0 1 1 1	0 0 0 0	7	0
		0 0 0 0	0 0 0 0	0	0

黒いマス=1 , 白いマス=0


図の数字をスペースキーに出すようにしてみましょう。

```
20 PATTERN C# & H 20 " 7 0 8.8.9 8.A 8.C 8.8.8.7 0 0 0 "
RUN
```

これでスペースキーを押してみてください。" 0 "が出ましたね。スペースに" 0 "が書き込まれたからです。

このように、文字を作ることが簡単に出来ます。グラフィック画面に出す絵も同じように書きます。


もう一度、使い方を整理します。

- ◎ C# テキストモード指定
- ◎ 文字コード 16進数の場合、& H 20 から& HFF までの数値で入力します。
 10進数の場合、32 から255 までの数値で入力します。
- ◎ 文字列式 8×8 のマス目ドットを黒く塗りつぶして文字や図形を描きます。
 黒い点は1 , 白い点は0 として各列毎に1 と0 を割当てます。
  は0 1 1 1 0 0 0 0 になります。これを中心
 から左右半分に分け、16 進数に変換します。
 0 1 1 1 は7 , 0 0 0 0 は0 で、" 70 " となります。

グラフィック画面指定

0から255までの番号が付けられます。

テキストモードと同じ使い方で入力します。



例

```
10 SCREEN 2,2:CLS
20 PATTERNS#0,"0103070F1F3F7FFF"
30 SPRITE 0,(10,0),0,1
RUN
```

▲ ← SCREEN 2へ出ます。

このような方法でキャラクターを作ってみてください。

パターンの描き方

パターン、テキストモードのとき 8×6

グラフィックモードのとき 8×8

で、画面上に表われます。もしテキストモードで 8×8 を指定しても右の 2 桁は無視されます。

まず、グラフ用紙を 8×8 のマス目に仕切り、ドット（マス目一つ分）を塗りつぶしてパターンを作ります。

塗った所を、1，空白を 0 とします。マス目の横に 0 や 1 の数字を並べて下さい。

並んだ 8 個の数字を真中から二つに分けて四桁づつにします。四桁の数字は 2 進数をあらわします。これを今度は 16 進数に直し 2 桁の 16 進数に直します。

10 進数、2 進数、16 進数の対照表を参考にして下さい。



16 進に直された 8 組の数字が文字列式として " " の中に代入されます。

文字は 8×6 、グラフィックは 8×8 で表現されます。


10進数	2進数	16進数
0	0 0 0 0	0
1	0 0 0 1	1
2	0 0 1 0 (けた上り)	2
3	0 0 1 1	3
4	0 1 0 0	4
5	0 1 0 1	5
6	0 1 1 0	6
7	0 1 1 1	7
8	1 0 0 0	8
9	1 0 0 1	9
10 (けた上り)	1 0 1 0	A
11	1 0 1 1	B
12	1 1 0 0	C
13	1 1 0 1	D
14	1 1 1 0	E
15	1 1 1 1	F
16	1 0 0 0 0	1 0 (けた上り)

MAG

PATTERN文でスプライト面に描かせる絵の大きさを指定する文です。




MAG 2



2 ビット × 2 ビットを
1 ドットとする。

MAG 3



MAG 2 を 4 つ合わせて
描く。

MAG 0 では、1 ビットを1 ドットとして、8 × 8 ドットの枠内で絵を描かせます。

MAG 1 では、8 ドット × 8 ドットのパターンの4 個組（# 0 ～# 3 ，# 4 ～# 7
，……，# 2 5 2 ～# 2 5 5 ）を結合して1 6 ドット × 1 6 ドットの枠
内で絵を描かせます。

MAG 2 では、2 ビット × 2 ビットを1 ドットとして、8 × 8 ドットの枠内で絵を描き
ます。
ビット数は1 6 ビット × 1 6 ビットになります。

MAG 3 では、MAG 2 で指定した枠を4 つ合わせて描くことができます。この場合は
3 2 ビット × 3 2 ビットになります。


SPRITE (スプライト)

スプライト機能を使う場合は、MAG 文、PATTERN 文、この SPRITE 文は絶対に欠かさないものです。

0~31 座標
SPRITE 面番号, (x, y), スプライト名称, 色


スプライト面は、0 番号から 31 番号までの 32 枚ありますから、何番の面に描くか指定します。面番号は 0 が 1 番前になり番号が大きくなるほど、後になります。スプライト同士が交差する時、番号が小さい方が前になります。

座標はグラフィック画面を使い、基準は MAG 命令で指定した枠の一番左上の座標です。



スプライト名称は、PATTERN文で付けた、S# 名称のことです。

PATTERN文で描いた絵にすき間があると、下の絵が見えます。これを利用して奥行のある立体感のある絵が作れます。



注 スプライトは、水平線上に最大4パターンまで表示されます。4個以上のパターンが並んだときは、優先順位の高い方の4個が表示されます。

第 6 章 関 数 そ の 2

コンピュータは、計算が得意です。計算機能を高めるために、三角関数を始め種々な関数が組み込まれております。

ABS (X)

機 能 式Xの絶対値を与えます。

形 式 ABS (X)

使い方 PRINT ABS (-5) CR 5

PRINT ABS (3 × (-6)) CR 18

RAD

機 能 角度からラジアンに変換します。

形 式 RAD (X)

使い方 0°, 15°, 30°, 45°, 60° をラジアンに変換する。

```
10 FOR I=0 TO 60 STEP 15
```

```
20 X=RAD ( I )
```

```
30 PRINT " RAD ( " ; I ; " ° ) = " ; X
```

```

40 NEXT I
RUN
RAD(0°)=0
RAD(15°)=.2617993878
RAD(30°)=.5235987756

```

DEG

機能 ラジアンから度に変換します。

形式 $\text{DEG}(X)$ X はラジアンです。

使い方 `PRINT DEG(0.26)` CR (14.896902673)

PI

機能 円周率を与えます。

形式 `PRINT PI` CR (3.1415926536)

使い方

```

10 INPUT "ハンケイ "; A
20 S=A^2*PI
30 PRINT "エン ノ メンセキ "; S
RUN

```

? 5

78.539816333

SIN (サイン)

機能 三角関数、正弦の値を与えます。

型式 SIN (X) 引数 (X) はラジアン

使い方

```
10 FOR TH=0 TO 90 STEP 30
```

```
20 S=SIN(RAD(TH))
```

```
30 PRINT TH;TAB(10);S
```

```
40 NEXT TH
```

```
RUN
```

```
0 0
```

```
30 .5
```

```
60 .86602540379
```

```
90 1
```

COS (コサイン)

機 能 三角関数、余弦の値を与えます。
形 式 COS (X) 引数 (X) ラジアン
使い方

```
10 FOR X=0 TO 90 STEP 30
20 A=COS ( RAD ( X ) )
30 PRINT X ; TAB ( 10 ) ; A
RUN
0          1
30         . 86602540379
60         . 500000000001
90         0
```

TAN (タンジェント)

機 能 三角関数、正弦の値を与えます。
形 式 TAN (X) 引数 (X) はラジアン
使い方

```
10 INPUT " ドスウ " ; A
```



```

20 X=TAN(RAD(A))
30 PRINT "TAN(" ; A ; "°) = " ; X
RUN

```

ドスウ 30
 TAN(30°) = .57735026919

ASN (アークサイン) 逆正弦

機 能 SIN θ の θ の値(度数)を求めます。
 形 式 ASN(X) Xは-1~1
 使い方

```

10 X=ASN(.5)
20 Y=DEG(X)
30 PRINT Y
RUN
30

```

ACS (アークコサイン) 逆余弦

機 能 COS θ の θ の値(度数)を求めます。

形 式 A C S (X) Xは-1 ~ 1
使い方

```
10 X=ACS(-1)
20 Y=DEG(X)
30 PRINT Y
RUN
180
```

A T N (アークタンジェント) 逆正接

機 能 逆正接の値を求めます。

形 式 A T N (X)

使い方 求められる値は $-\frac{\pi}{2}$ から $\frac{\pi}{2}$ の間の値です。

```
10 X=ATN(1)
20 PRINT X
RUN
.7853981634
```

SGN (サイン)

機 能 数値の符号を求める関数です。

 Xの値が負の時は - 1

 0の時は 0

 正の時は 1

形 式 SGN (X)

使い方

```
10 FOR I=-2 TO 2
20 N=SGN ( I )
30 PRINT "SGN ( " ; I ; " ) = " ; N
40 NEXT I
RUN
SGN ( - 2 ) = - 1
SGN ( - 1 ) = - 1
SGN ( 0 ) = 0
SGN ( 1 ) = 1
SGN ( 2 ) = 1
```

LOG

機 能 e を底とする対数を求めます。(自然対数)

形 式 LOG (X)

使い方

```
10 FOR J=1 TO 3
20 X=LOG ( J ) ← 引数 J は正の値です。
30 PRINT " LOG ( " ; J ; " ) = " ; X
40 NEXT J
RUN
LOG ( 1 ) = 2 . 67468532E-11
LOG ( 2 ) = . 69314718057
LOG ( 3 ) = 1 . 0986122886
```

LTW

機 能 2 を底とする対数を求めます。

形 式 LTW (X)

使い方 LOG と同じです。

LGT

機能 10を底とする対数を求めます。(常用対数)
形式 LGT(X)
使い方 10, 100, 1000の常用対数を求めます。

```
10 N=1
20 N=N*10
30 X=LGT(N)
40 PRINT "LGT(" ; N ; ")=" ; X
50 IF N<1000 THEN 20
RUN
LGT(10)=1
LGT(100)=2
LGT(1000)=3
```

EXP

機能 e のべき乗を求めます。
形式 EXP(X)
使い方 e^1 、 e^2 、 e^3 をそれぞれ求めてみます。

```

10 FOR I=1 TO 3
20 X=EXP(I)
30 PRINT "EXP( "; I; " ) "; X
40 NEXT I
RUN
EXP(1)=2.7182818284
EXP(2)=7.3890560987
EXP(3)=20.085536923

```

SQR

機能 数値の平方根を求めます。
 形式 SQR(X)
 使い方 $\sqrt{2}$, $\sqrt{3}$, を求めてみます。

```

10 INPUT "スウジ "; A
20 X=SQR(A)
30 PRINT "ルート "; A; " = "; X
40 GOTO 10
RUN

```

スウジ 2

ルート=1 . 4 1 4 2 1 3 5 6 2 4

スウジ 3

ルート=1 . 7 3 2 0 5 0 8 0 7 6

HEX\$

機能 10進数の数値を16進に変換します。

形式 HEX\$(X)

使い方 16進数に変換出来る数値の範囲は-32768~32767の間です。
-10, -5, 0, 5, 10, 15を16進に変換します。

```
10 FOR S=-10 TO 15 STEP 5
```

```
20 X$=HEX$(S)
```

```
30 PRINT S;"=";X$
```

```
40 NEXT S
```

```
RUN
```

```
-10=FFFF6
```

```
-5=FFFB
```

```
0=0          10=A
```

```
5=5          15=F
```

SOUND

SC-3000には、シンセサイザ機能があります。

例

SOUND 1, 1000, 15



チャンネル 周波数 音量

1000ヘルツの音が出ました。

<チャンネル>

1つのチャンネルからは、1つの音しか出ません。チャンネルは、0～5までの6個の指定があります。（三重和音まで発生できます。）

0 : 音を消します。

例 SOUND 0

1～3 : 110Hz～の音を出します。

4 : ホワイトノイズの選択

5 : 同期ノイズの選択

ノイズといっても、雑音とは違います。
主に効果音に使われます。

< 周 波 数 >

チャンネル1～3の時は、周波数(Hz)を書きます。

チャンネル指定が、4又は5のときは、

0～2 : 三段階の決定されている周波数になります。

3 : チャンネル3で指定する周波数になります。

< 音 量 >

0 : 音を消す。

1 : 最小の音量。

↓

15 : 最大の音量。

これによって、ゲーム等の効果音や、次表に依りメロディを奏でることができます。

周 波 数 表

音 階			f1	f2	f3	f4	f5
C	ド	ハ		131	262	523	1047
C \sharp , D \flat				139	277	554	1109
D	レ	ニ		147	294	587	1175
D \sharp , E \flat				156	311	622	1245
E	ミ	ホ		165	330	659	1319
F	ファ	ヘ		175	349	698	1397
F \sharp , G \flat				185	370	740	1480
G	ソ	ト		196	392	784	1568
G \sharp , A \flat				208	415	831	1661
A	ラ	イ	110	220	440	880	1760
A \sharp , B \flat			117	233	466	932	
B	シ	ロ	123	247	494	988	

周波数単位. Hz.

B E E P

プログラムの中で、短い音を出したいときに使います。

B E E P ビッと音が鳴る。

B E E P 0 B E E P 音をとめる。

B E E P 1 ビーと鳴り続ける。

B E E P 2 ビボビボと鳴る。

例

```
10 A$="SEGA PERSONAL COMPUTER"  
20 FOR I=1 TO LEN(A$)  
30 PRINT MID$(A$,I,1);  
40 BEEP  
50 FOR J=0 TO 100:NEXT J,I  
60 END
```

I N P

- I/O エリアの内容を読みだします。O U Tと逆の機能を持ちます。
I/O ポート番号を指定して、ポート上にあるデータを読み出します。
使い方

```
10 A=INP(&HBE)
20 PRINT A
RUN
32
```


行番号10でI/O ポート番号B E (16進数) にあるデータを変数Aに読み出します。
この場合、コンピュータの状態によって結果が変わることがあります。
I/O ポート番号はシステム中ですでに決定されているもので、0 ~ 255 (& H 0 0 ~
& H F F) までの整数です。
外部入力装置であるジョイスティックなどの状態を知るときなどは、この命令によって知ることが出来ます。

STICK (n)

パラメタ n : 1 = ジョイスティック 1

2 = ジョイスティック 2

(戻す値)



STRIG (n)

パラメタ n : 1 : ジョイスティック 1

2 : ジョイスティック 2

(戻す値)

0 : オフ

1 : トリガ (左) ON

2 : トリガ (右) ON

3 : トリガ (左, 右) ON

<STICK,STRIG>

接続したJOYSTICKの状態を知るプログラム

```
10 REM JOY STICK TEST
20 B$="SHOOT":CLS
30 P1=STICK(1):P2=STICK(2)
40 S1=STRIG(1):S2=STRIG(2)
50 F1$=" ":F2$=" "
60 IF P1=1 THEN F1$="UP"
70 IF P1=3 THEN F1$="RIGHT"
80 IF P1=5 THEN F1$="DOWN"
90 IF P1=7 THEN F1$="LEFT"
100 IF P2=1 THEN F2$="UP"
110 IF P2=3 THEN F2$="RIGHT"
120 IF P2=5 THEN F2$="DOWN"
130 IF P2=7 THEN F2$="LEFT"
140 IF S1>0 THEN F1$=F1$+B$+STRING$(S1)
150 IF S2>0 THEN F2$=F2$+B$+STRING$(S2)
160 CURSOR 1,10:PRINT"PLAYER 1";F1$
170 CURSOR 1,15:PRINT"PLAYER 2";F2$
180 GOTO 20
```

VRAM MAP

VRAM (16Kバイト)

& H0000

グラフィックⅡモード
パターンゼネレータ
テーブル
(6144 バイト)

表示がテキストの時

テキストモード
パターンゼネレータテーブル
(2048 バイト)

& H1800

注 1

& H2000

グラフィックⅡモード
カラーテーブル
(6144 バイト)

表示がグラフィックの時

スプライトゼネレータテーブル
(2048 バイト)

& H3800

グラフィックⅡモードパターン
名称テーブル

(768 バイト)

& H3B00

スプライト属性テーブル

(128 バイト)

(空)

& H3C00


テキストモードパターン
名称テーブル

(960 バイト)

(空)

注1 &H1800～&H1FFFの2Kバイトは表示モード（テキスト／グラフィック）によってテーブル内容が異なる。
選ばれてない側のモードのテーブル内容はメインメモリ（RAM）にセーブされます。

VPOKE アドレス, アスキーデータ. (テキストモード)



VRAM MAP 1 部

テキスト画面のアドレス計算は次のように行なわれます。

アドレス(テキスト) = $y * 40 + x + \&H3C00$

ただし ($x = 0 \sim 39$, $y = 0 \sim 23$)

送り出すデータはその文字のアスキーコード(10進数で0~255又は16進で0~&HFF)である。

注

前図左のように通常のテキスト画面とは横軸が2カラムづれているために、**CURSOR**文を使用した表示位置と**VPOKE**による表示位置とは横方向に2つ程づれます。

142 ページ参照

グラフィックのアドレス計算は次のように行なわれる。

$$\text{グラフィックアドレス} = \text{INT}(y/8) * 256 + \text{INT}(x/8) * 8 + y \text{ MOD } 8$$

ただし (y は 0 ~ 191, x は 0 ~ 255)

上の計算によって得られるアドレスは指定された場所の横方向 8 ビット (ドット) の先頭アドレスです。指定したアドレスは、その先頭アドレスより左から $x - \text{INT}(x/8) * 8$ ビット目である。

送り出すデータは、横 1 列のビットパターンで表示される 16 進、又は 10 進値である。

例  ⇒ &H93 (147)

同様に、グラフィック色指定の為のカラーテーブルアドレスは、上記のアドレスに &H 2000 を加えたものであり、送り出すデータは 1 B (バイト) 長の自然数 (0 ~ 255) で、2 進データに直して上位 4 bit が指定色番号で下位 4 bit が背景色番号です。
(グラフィックのパターンゼネレータテーブルとカラーテーブルのアドレスは 1 対 1 に対応している)

グラフィックカラーテーブルアドレス

$$\begin{aligned} &= \text{INT}(y/8) * 256 + \text{INT}(x/8) * 8 \\ &\quad + y \text{ MOD } 8 + \&H2000 \end{aligned}$$


yは0～191

xは0～255

色データ=指定色番号*16+背景色番号
 (0～15) (0～15)

SCREEN

表示画面



VPEEK

VPEEK用法はVPOKEのアドレスを参考に使用してください。

VRAM内パターンゼネレータテーブルの内容を読み取るプログラム

例

```

10 AD=&H1800+&H31*8      : REM " 1 "のパターンの
20 FOR A=AD TO AD+7        先頭アドレス
30 DA=VPEEK(A)
40 PRINT HEX$(DA)
50 NEXT A

```

20		1				
60	1	1				
20		1				
20		1				
20		1				
20		1				
70	1	1	1			
00						

変数・配列

数 値 変 数 A, B, Z, A A, A B, Z Z

A 0, A. 1 A 9

数 値 配 列 (添字,) 3 次元まで A (1 5), B (5, 5),

A C (3, 3, 3)

文 字 列 変 数 A \$, A B \$, A 1 \$

文 字 列 配 列 (添字,) 3 次元まで A \$ (1 5),

B \$ (5, 5), A C \$ (3, 3, 3)

- ・ 変数名は、先頭の 1 文字が英字、以降が英字または数字とする。長さは何文字でも良いが先頭の 2 文字で区別する。
- ・ 変数と配列の名前が同じでも良い。

<数値変数、配列の範囲>

± 9 . 9 9 9 9 9 9 9 9 9 9 E - 9 9

}

± 9 . 9 9 9 9 9 9 9 9 9 9 E + 9 9

<文字列変数、配列の範囲>

文字長さ 0 ~ 2 5 5

定 数

<数値定数>

- 整数形式 (例) 3, -2, 99926768
- 小数点形式 (例) 0.2, .3, -5.3, 86.0
- 指数形式 (例) 3, E99, -6E3, 0.3E+5,
4E-82
- 16進形式 &H 16進の値 0000 ~ FFFF
(例) &H64 100 と同じ
&HFFFF -1 と同じ

<文字列定数>

" で囲む、" は " を 2 つ用いて示す。

- (例) "ABC" → 文字 ABC
" " → 文字 NULL (なし)
" " " " → 文字 "
"A3" "64" → 文字 A3"64

内 容	制 限
画面から内部に取り込まれる文字数	256 文字
ラインバッファから予約語におとして実際のテキストイメージにできる文字数	256 文字
文字列として扱うことの可能な文字数	255 文字
演算の優先順位等のレベル数	32 レベル
文字列演算のためのエリア	300 文字
FOR ~ NEXT のネスティングのレベル数	16 レベル
GOSUB, RETURN のネスティング レベル数	8 レベル












キャラクタセット

	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F
0				0	@	P	\	p		■		-	タ	ミ	■	火
1			!	1	A	Q	a	q	—	×	。	ア	チ	ム	■	水
2			"	2	B	R	b	r	+	+	「	イ	ツ	メ	■	木
3			#	3	C	S	c	s	—	/	」	ウ	テ	モ	■	金
4			\$	4	D	T	d	t	+	\	、	エ	ト	ヤ		土
5			%	5	E	U	e	u	+	▲	。	オ	ナ	ユ	■	♠
6			&	6	F	V	f	v	「	▲	ツ	カ	ニ	ヨ		♥
7			'	7	G	W	g	w	「	▲	ア	キ	ヌ	ラ	—	♦
8			(8	H	X	h	x	「	▲	イ	ク	ネ	リ	■	♣
9)	9	I	Y	i	y	「	—	ウ	ケ	ノ	ル	■	☺
A			*	:	J	Z	j	z	「	—	エ	コ	ハ	レ	■	🚗
B			+	;	K	[k	{	「	—	オ	サ	ヒ	ロ	○	I
C			,	<	L	¥	l	:	「	—	ヤ	シ	フ	ワ	●	H
D			-	=	M]	m	}	「	—	ユ	ス	ヘ	ン	年	👤
E			.	>	N	^	n	~	↑	—	ヨ	セ	ホ	。	月	÷
F			/	?	O	π	o		←		ッ	ソ	マ	。	日	

コントロールコード

キャラクターコード

32	SP	48	0	64	Ⓐ	80	P	96	／	112	p	128	□	144	■
33	!	49	1	65	A	81	Q	97	a	113	q	129	▢	145	⊠
34	*	50	2	66	B	82	R	98	b	114	r	130	▣	146	⊡
35	#	51	3	67	C	83	S	99	c	115	s	131	▤	147	⊢
36	\$	52	4	68	D	84	T	100	d	116	t	132	▥	148	⊣
37	%	53	5	69	E	85	U	101	e	117	u	133	▦	149	⊤
38	&	54	6	70	F	86	V	102	f	118	v	134	▧	150	⊥
39	▼	55	7	71	G	87	W	103	g	119	w	135	▨	151	⊦
40	(56	8	72	H	88	X	104	h	120	x	136	▩	152	⊧
41)	57	9	73	I	89	Y	105	i	121	y	137	▪	153	□
42	*	58	:	74	J	90	Z	106	j	122	z	138	▫	154	▬
43	+	59	;	75	K	91	[107	k	123	{	139	▬	155	▭
44	,	60	<	76	L	92	¥	108	l	124	•	140	▬	156	▮
45	-	61	=	77	M	93]	109	m	125	}	141	▬	157	▯
46	.	62	>	78	N	94	∧	110	n	126	〰	142	↑	158	▰
47	/	63	?	79	O	95	π	111	o	127		143	↩	159	▱

160		176	ー	192	タ	208	ミ	224		240	火
161	。	177	ア	193	チ	209	ム	225		241	水
162	「	178	イ	194	ツ	210	メ	226		242	木
163	」	179	ウ	195	テ	211	モ	227		243	金
164	、	180	エ	196	ト	212	ヤ	228		244	土
165	。	181	オ	197	ナ	213	ユ	229		245	♠
166	ヲ	182	カ	198	ニ	214	ヨ	230		246	♥
167	ア	183	キ	199	ヌ	215	ラ	231		247	♦
168	イ	184	ク	200	ネ	216	リ	232		248	♣
169	ウ	185	ケ	201	ノ	217	ル	233		249	☹
170	エ	186	コ	202	ハ	218	レ	234		250	☂
171	オ	187	サ	203	ヒ	219	ロ	235		251	⌈
172	ヤ	188	シ	204	フ	220	ワ	236		252	⌋
173	ユ	189	ス	205	ヘ	221	ン	237	年	253	人
174	ヨ	190	セ	206	ホ	222		238	月	254	÷
175	ツ	191	ソ	207	マ	223		239	日	255	

コマンド、ステートメント、関数

コマンド

No.	コマンド	機能
1	LIST	プログラムを画面に表示する。
2	LLIST	プログラムをプリンタに印字する
3	SAVE	プログラムをカセットテープに録音する。
4	VERIFY	メモリ内のプログラムとカセットテープに録音されたプログラムの比較。
5	LOAD	カセットテープのプログラムをメモリにロードする。
6	RUN	プログラムを実行する。
7	CONT	中断されていたプログラムを続行する。
8	NEW	変数、プログラムをクリアする。
9	DELETE	プログラムを部分的にクリアする。
10	AUTO	ラインナンバを自動発生する。
11	RENUM	ラインナンバをつけなおす。

NO.	ステートメント	機 能
1	REM	コメント
2	STOP	プログラムの一時中断、CONT により続行。
3	END	プログラムの実行終了。
4	LET	代入（省略可）。
5	PRINT 又は ?	ディスプレイに表示する。
6	LPRINT 又は L ?	プリンターに印字する。
7	INPUT	キー入力
8	READ	“DATA” ステートメントのデータを読みとる。
9	DATA	“READ” ステートメントにより読み込まれるデータを示す。
10	RESTORE	“READ” ステートメントにより読み込む “DATA” ステートメントの場所を指定する。
11	DIM	配列の宣言。
12	ERASE	宣言された配列をクリアする。
13	DEFFN	ユーザー関数を定義する。
14	GOTO	分岐。
15	GOSUB	サブルーチンへの分岐。
16	RETURN	サブルーチンからの戻り。
17	ON ~ GOTO ON ~ GOSUB	分岐する行番号を選択する。 分岐する行番号を選択する。
18	FOR ~ TO STEP ~	NEXT ステートメントとの間を指定された条件で繰り返す。 STEP は省略化。
19	NEXT	“FOR” ステートメントによる繰り返しの場所を指定。
20	IF ~ THEN	条件分岐。
21	CONSOLE	画面スクロールの範囲、クリック音、キャラクターセットを指定する。

NO.	ステートメント	機 能
22	CLS	画面をクリアする。
23	SCREEN	画面の切替え。
24	COLOR	色の指定
25	PATTERN	文字、スプライトのパターンを変更する。
26	CURSOR	表示位置の指定。
27	POSITION	座標の指定。
28	PSET	点をうつ。
29	PRESET	点で消す。
30	LINE	線を描く。
31	BLINE	線で消す。
* 32	CIRCLE	円を描く。
* 33	BCIRCLE	円で消す。
34	PAINT	かこまれた範囲をぬりつぶす。
* 35	SPRITE	スプライトの位置、色、パターンを指定。
* 36	MAG	スプライトの大きさを指定する。
37	SOUND	効果音を発生させる。
38	BEEP	BEEP 音を発生させる。
39	HCOPY	テキスト画面の内容をプリンタにコピーする。
* 40	CALL	機械語 サブルーチンへ分岐する。
* 41	POKE	メモリーへの書き込み。
* 42	OUT	出力ポートに出力する。
* 43	VPOKE	VRAM にデータを書き込む。

*印のステートメントは BASIC レベル 2 では使えません。

関 数

NO.	関 数	内 容
1	ABS(x)	x の絶対値を求める。
2	RND(x)	乱数を作る。
3	SIN(x)	x のサインを求める。
4	COS(x)	x のコサインを求める。
5	TAN(x)	x のタンジェントを求める。
6	ASN(x)	x のアークサインを求める。
7	ACS(x)	x のアークコサインを求める。
8	ATN(x)	x のアークタンジェントを求める。
9	LOG (x)	x の自然対数を求める。
10	LGT (x)	x の常用対数を求める。
11	LTW(x)	x の 2 を底とする対数を求める。
12	EXP(x)	e の x 乗を求める。
13	RAD(x)	度をラジアン単位に変換する。
14	DEG(x)	ラジアンを度単位に変換する。
15	PI	円周率を求める。
16	SQR(x)	x の平方根を求める。
17	INT(x)	x を越えない最大の整数を求める。
18	SGN(x)	x の正負符号を与える。
19	ASC(s)	文字列 s の最初のコードを数値で与える。
20	LEN(s)	文字列 s が何文字あるかを与える。
21	VAL(s)	文字列 s を数値に変換する。

NO.	関 数	内 容
22	CHR\$(x)	x の対応する文字や機能を与える。
23	HEX\$(x)	x の 16 進数文字列を与える。
24	INKEY\$(x)	キーボードが押されたかどうか調べる。キーボードが押されればその文字を与える。押されなければ 0 (NULL)
25	LEFT\$(s,x)	文字列 s の左から x 番目までの文字列を与える。
26	RIGHT\$(s,x)	文字列 s の右から x 番目までの文字列を与える。
27	MID\$(s,x,y)	文字列 s の右から x 番目から長さ y の文字列を与える。 y は省略可です。その時は最後まで与える。
28	STR\$(x)	x をそれを表示する文字列に変換する。
29	TIME\$	内部クロックの時刻を与える。
* 30	PEEK(x)	メモリー x 番地の内容を与える。
* 31	INP(x)	入力ポートの入力内容を与える。
32	FRE	ユーザー用メモリーエリアの空きを与える。
33	SPC(x)	プリントステートメントで使用、スペースをあける。
34	TAB(x)	プリントステートメントで使用、表示位置の指定。
* 35	STICK(n)	ジョイスティック n の方向を示す。
* 36	STRIG(n)	ジョイスティック n のトリガーボタンの状態を示す。
* 37	VPEEK(x)	VRAM x 番地の内容を与える。

エラーメッセージ

1 表示 FORMAT

- (1) コマンド入力又はステートメントをダイレクトに入力した時、エラーが発生した場合。

? メッセージ error

- (2) テキストを実行中にエラーが発生した場合。

? メッセージ error in ラインナンバー

- (3) INPUT ステートメントによるキー入力データに誤りがあった場合。

? メッセージ

メッセージ	内 容
System	BASIC インタプリタのプログラムの動作エラー。 (普通はありえない)
N-formula too Complex	数値式が複雑すぎる。
S-formula too Complex	文字列式が複雑すぎる。
Overflow	値や演算結果が許容範囲を越えた。
Division by Zero	除算の分母が 0 。
Function Parameter	関数のパラメタが異常。
String too long	文字列の長さが 255 をオーバーした。
Stack overflow	・カッコの使いすぎ。 ・ PAINT する図形が複雑すぎる。 ・ ユーザ定義関数が自分自身を呼び出している。
Out of memory	メモリが足りない。 ・ テキスト ・ 変数 ・ 配列
Number of Subscripts	添字の数 (偶数) が異常。
Value of Subscript	添字の値がおかしい。
Syntax	文法上のエラー。
Command Parameter	コマンドのパラメタが異常。
Line number over	AUTO 又は RENUM においてラインナンバーが 65535 オーバーする。
Illegal line number	ラインナンバーがおかしい。
Line image too long	ラインが長くなりすぎる。(RENUM 等)

メッセージ	内 容
Undefined line number	指定されたライン番号がない。(RENUM , GOTO , GOSUB , IF-THEN , RESTORE , RUN)
Type mismatch	代入する側と代入される側のタイプが合わない。 (数値、文字列)
Out of DATA	READ ステートメントにより読み込もうとしたが、DATA 文のデータがない。
RETURN without GOSUB	GOSUB を行なわないで RETURN 文が実行された。
GOSUB nesting	GOSUB の入れ子が 16 レベルをこえた。
NEXT without FOR	NEXT に対応する FOR ステートメントが無い。
FOR nesting	FOR ~ NEXT の入れ子が 8 レベルをこえた。
Statement Parameter	ステートメントのパラメタが異常。
Con't continue	CONT コントによる続行が不可能。
FOR variable name	FOR ステートメントのループ変数が数値変数でない。 (文字型又は配列)
Array name	DIM ステートメントのパラメタが配列でない。
Redimensioned array	配列を 2 重に定義しようとした。
Undefined array	未定義の配列を ERASE しようとした。
No program	プログラムがテキスト中に無いのに SAVE しようとした。
Memory writing	メモリへの書き込みエラー。(LOAD 時)

メッセージ	内 容
Device not ready	プリンタが接続されていない、又は故障。
Undefined function	定義されていないユーザ関数を呼び出した。
Verifying	テープのプログラムとの比較エラー。
Function buffer full	9 個以上のユーザー定義関数を定義しようとした。
Illegal direct	ダイレクトステートメント実行不可。
Redo from start	INPUT ステートメント入力データがおかしい。 ・最初から入力しなおしを要求。
Extra ignored	INPUT ステートメント入力データがおかしい。 ・余分なデータが入力された。 ・余分なデータは無視。
Unprintable	上記以外のエラー。

サンプル プログラム

格 子 P A I N T

```
10 SCREEN 2,2:CLS
50 X=10:Y=10:XX=250:YY=190
100 REM タテ LINE
110 FOR I=1 TO 12
120 LINE (X,Y)-(X,YY),1
130 X=X+20
140 NEXT I
200 REM 横 LINE
210 X=0
220 FOR I=1 TO 10
230 LINE (X,Y)-(XX,Y),1
240 Y=Y+20
250 NEXT I
300 REM PAINT
310 A=RND(1)*240
320 B=RND(1)*185
330 C=RND(1)*15
340 PAINT(A,B),C
350 GOTO 300
```

LINE

```
10 SCREEN 2,2:CLS:COLOR,15
15 FOR I=0 TO 30
20 A=INT(RND(1)*16)
30 B=INT(RND(1)*128):C=INT(RND(1)*96)
40 D=INT(RND(1)*256):E=INT(RND(1)*192)
50 LINE(B,C)-(D,E),A,B
55 NEXT I
100 GOTO 10
```

LINEBF

```
10 SCREEN 2,2:CLS
20 A=INT(RND(1)*16)
30 B=INT(RND(1)*128):C=INT(RND(1)*96)
40 D=INT(RND(1)*256):E=INT(RND(1)*192)
50 LINE(B,C)-(D,E),A,BF
100 GOTO 20
```

PAINT

```
10 SCREEN 2,2:CLS
20 FOR I=0 TO 255 STEP 16
30 LINE(I,0)-(I,191):NEXT I
40 FOR I=0 TO 191 STEP 16
50 LINE(0,I)-(255,I):NEXT I
60 C=INT(RND(1)*16)
70 X=INT(RND(1)*256):Y=INT(RND(1)*192)
80 PAINT(X,Y),C
90 GOTO 60
```


CIRCLE 1

```
10 SCREEN2,2:CLS:COLOR15
20 FOR R=1TO96
30 C=INT(RND(1)*16)
40 CIRCLE(128,96),R,C,1,0,1,
50 NEXT R
60 GOTO 20
```

CIRCLE 2

```
10 SCREEN2,2:CLS:COLOR15
20 FOR R=1TO90 STEP 5
30 C=INT(RND(1)*16)
40 CIRCLE(128,96),R,C,1,0,1,
50 NEXT R
60 PAINT(0,0),5
```

CIRCLE 3

```
10 SCREEN2,2:CLS
20 X=INT(RND(1)*256)
30 Y=INT(RND(1)*192)
40 C=INT(RND(1)*16)
50 R=INT(RND(1)*20)
60 CIRCLE(X,Y),R,C,1,0,1
70 GOTO 20
```

BCIRCLE

```
10 SCREEN2,2:CLS:COLOR15
20 FOR R=1 TO 96 STEP 10
30 C=INT(RND(1)*16)
40 CIRCLE(128,96),R,C,1,0,1
50 NEXT R
60 FOR I=91 TO 1 STEP -10
70 BCIRCLE(128,96),I,,1,0,1
80 NEXT I
100 GOTO 20
```

スプライト サンプル プログラム

```
10  M=1
20  SCREEN 2,2:CLS
30  MAG M:C=RND(1)*13+1
40  CURSOR 10,10:PRINT CHR$(17);"MAG";M
50  FOR Y=0 TO 191 STEP 4
60  PATTERN S#0,"00193F3C1C0D0F7B"
70  PATTERN S#1,"0C0F0F0F07031B07"
80  PATTERN S#2,"00CCFE9E9CD878EC"
90  PATTERN S#3,"1AFAF8F0EC7C3800"
100 Y1=Y:GOSUB 200
110 PATTERN S#0,"00193F3C1C0D0F1B"
120 PATTERN S#1,"2C2F0F071B1F0E00"
130 PATTERN S#2,"00CCFE9E9CD878EF"
140 PATTERN S#3,"18F8F8F8F0606C70"
150 Y1=Y+2:GOSUB 200
160 NEXT Y
170 M=M+2:IF M>3 THEN M=1
180 GOTO 20
200 SPRITE 0,(120,Y1),0,C
210 SPRITE 0,(120,Y1+1),0,C
220 RETURN
```

ジャンプ

ゲームに応用できるサンプルです。

  キーで左右に動きます。

 キーで 小さくジャンプ

 キーで 大きくジャンプ

```
10 CLS:X=20:Y=15
20 CURSOR X,Y:PRINT "A"
30 Z$=INKEY$
40 IF Z$=CHR$(28) THEN E=1:GOSUB 100
50 IF Z$=CHR$(29) THEN E=-1:GOSUB 100
60 IF Z$=CHR$(12) THEN E=-2:GOSUB 200
70 IF Z$=CHR$(8) THEN E=-6:GOSUB 200
80 GOTO 30
100 CURSORX,Y:PRINT" "
110 FOR N=0 TO 10:NEXT N
115 X=X+E:IF X=0 OR X=37 THEN X=X-E
120 CURSORX,Y:PRINT"A"
130 RETURN
```

```

200 CURSORX,Y:PRINT"
210 Y=Y+E
220 CURSORX,Y:PRINT"A"
230 FOR N=0 TO 50:NEXT
240 CURSORX,Y:PRINT" "
250 Y=Y-E
260 CURSORX,Y:PRINT"A"
280 RETURN

```

スロット・マシン

画面中央に 3 つの数字が並びます。

左から同じ数字が	2 つ並ぶと	カケテン × 2 倍
	3 つ並ぶと	カケテン × 4 倍

```

10 T=100
20 CLS
30 PRINT"ナンテン カケマスカ ? モチテン ";T;"テン"
40 INPUT K:PRINT:PRINT"カケテン";K;"テン"
50 FOR N=0 TO 9
60 A=INT(RND(1)*3+1)

```

```

70 B=INT(RND(1)*3+1)
80 C=INT(RND(1)*3+1)
90 CURSOR10,8:PRINT A;B;C
100 NEXT
110 IF A=B THEN 140
120 PRINT"ﾄｽﾞレ":T=T-K:GOTO 170
130 T=T-K:GOTO 20
140 IF A=C THEN 160
150 PRINT"ﾌﾀﾘ":T=T+K*2:GOTO 170
160 PRINT"ｽｽﾌﾀﾘ":T=T+K*4:GOTO 170
170 IF T<=0 THEN PRINT"ﾄｯﾌﾟ ﾀﾞｽ":END
180 FOR W=0 TO 300:NEXT W:GOTO 20

```

一筆書き

連続した線を描く場合、LINE-(X,Y),色 が使えます。いろいろな応用をしてみましょう。

```

5 SCREEN 2,2:CLS
10 LINE(80,150)-(80,100),1
20 LINE-(120,70),2
30 LINE-(160,100),4

```

```
40 LINE-(160,150),6
50 LINE-(80,100),7
60 LINE-(160,100),8
70 LINE-(80,150),10
80 LINE-(160,150),13
90 COLOR,9,(80,100)-(159,150),5
100 PAINT(120,80),1
110 PAINT(120,110),7
120 PAINT(120,140),3
140 C=C+1:IF C=15 THEN C=0
150 COLOR,C,(81,100)-(159,150),0
160 FOR W=0 TO 220:NEXT W
170 GOTO 140
```

円

円を描く場合には、SIN、COS で描くことができます。円の大きさは、行番号 50 と 60 の数値を変えて下さい。

円を描く点と点の間は行番号 30 の STEP の数値を .1 にすると間が詰まります。

```
10 SCREEN 2,2:CLS
20 POSITION(120,90),0,0
30 FOR R=0 TO 50 STEP .5
40 S=R/50*PI*2
50 X=INT(SIN(S)*50)
60 Y=INT(COS(S)*50)
70 PSET(X,Y),1
80 NEXT R
90 GOTO 90
```


お わ り に

BASIC レベル 3 をお使いになりましたいかがでしたか。レベル 3 はグラフィック命令や関数も多く、使いやすく充実した機能を持たせたので、工夫次第で楽しい使い方が出来たと思います。

ページ数の都合で簡単な内容になりましたが、サンプルのプログラムの数値を変えたり、命令語を組合せて応用して下さい。始めはエラーばかり出ると思いますが、あきらめずに LIST を見直してプログラムして下さい。

BASIC 言語の書籍や雑誌が多数出版されておりますので、参考にされると良いと思います。メーカーによって、BASIC の命令文に違いがありますが、基本言語はほとんど同じです。PEEK, POKE 以外同じ言語であれば試しに入力してみてください。

BASIC のプログラミングが、楽しいホビーになると思います。

以 上

SC-3000 BASIC レベル 3 解説書

発行所 株式会社 セガエンタープライゼス
パーソナルコンピュータ事業本部

〒 144 東京都大田区羽田 1 丁目 2 番 12 号
☎ 03 - 742 - 3171

昭和 59 年 7 月 15 日 発行

印刷・製本 (株) 日本現図社

〒 140 東京都品川区北品川 1 丁目 29 番 11 号
☎ 03 - 471 - 3353

© SEGA 1983

無断転載複製を禁ず
落丁・乱丁本はお取替え致します。

株式会社 **セガ**・エンタープライゼス